

The Topology of Lazy Lists

Stefan Friedrich

December 12, 2009

Abstract

This directory contains two theories. The first, `Topology`, develops the basic notions of general topology. The second, `LList.Topology`, develops the topology of lazy lists.

Contents

1	A bit of general topology	2
1.1	Preliminaries	2
1.2	Definition	2
1.3	Neighbourhoods	9
1.4	Closed sets	11
1.5	Core, closure, and frontier of a set	13
1.5.1	Core	13
1.5.2	Closure	14
1.5.3	Frontier	15
1.5.4	Adherent points	16
1.6	More about closure and core	18
1.7	Dense sets	19
1.8	Continuous functions	20
1.9	Filters	24
1.10	Convergence	30
1.11	Separation	32
1.11.1	T0 spaces	32
1.11.2	T1 spaces	33
1.11.3	T2 spaces (Hausdorff spaces)	34
1.11.4	T3 axiom and regular spaces	38
1.11.5	T4 axiom and normal spaces	39
2	The topology of llists	40
2.1	The topology of all llists	40
2.2	The topology of infinite llists	43
2.3	The topology of non-empty llists	48

1 A bit of general topology

```
theory Topology
imports FuncSet Zorn
begin
```

This theory gives a formal account of basic notions of general topology as they can be found in various textbooks, e.g. in [2] or in [3]. The development includes open and closed sets, neighbourhoods, as well as closure, open core, frontier, and adherent points of a set, dense sets, continuous functions, filters, ultra filters, convergence, and various separation axioms.

We use the theory on “Pi and Function Sets” by Florian Kammüller and Lawrence C Paulson.

1.1 Preliminaries

```
lemma seteqI:
```

```
"[[  $\bigwedge x. x \in A \implies x \in B$ ;  $\bigwedge x. x \in B \implies x \in A$  ]]  $\implies A = B$ "
by auto
```

```
lemma subset_mono: " $A \subseteq B \implies M \subseteq A \longrightarrow M \subseteq B$ "
by auto
```

```
lemma diff_diff:
```

```
" $C - (A - B) = (C - A) \cup (C \cap B)$ "
by blast
```

```
lemma diff_diff_inter: " $[B \subseteq A; B \subseteq X] \implies (X - (A - B)) \cap A = B$ "
by auto
```

```
lemmas diffsimps = double_diff Diff_Un vimage_Diff
Diff_Int_distrib Diff_Int
```

```
lemma vimage_compose:
```

```
" $f : A \rightarrow B \implies A \cap (f^{-1} B \cap f^{-1} g^{-1} m) = A \cap (g \circ f)^{-1} m$ "
by (auto dest: funcset_mem)
```

```
lemma funcset_comp:
```

```
"[[  $f : A \rightarrow B$ ;  $g : B \rightarrow C$  ]]  $\implies g \circ f : A \rightarrow C$ "
by (auto intro!: funcsetI dest!: funcset_mem)
```

1.2 Definition

A topology is defined by a set of sets (the open sets) that is closed under finite intersections and infinite unions.

```
types
```

```
'a top = "'a set set"
```

```
definition
```

```
carr :: "'a top  $\Rightarrow$  'a set" ("carrier $\iota$ ") where
" carr T =  $\bigcup T$ "
```

definition

```
is_open :: "'a top  $\Rightarrow$  'a set  $\Rightarrow$  bool" ("_ openz" [50] 50) where
  "is_open T s  $\longleftrightarrow$  s  $\in$  T"
```

locale carrier =

```
  fixes T :: "'a top" (structure)
```

lemma (in carrier) openI:

```
  "m  $\in$  T  $\Longrightarrow$  m open"
  by (simp add: is_open_def)
```

lemma (in carrier) openE:

```
  "[[ m open; m  $\in$  T  $\Longrightarrow$  R ]  $\Longrightarrow$  R"
  by (auto simp: is_open_def)
```

lemma (in carrier) carrierI [intro]:

```
  "[[ t open; x  $\in$  t ]  $\Longrightarrow$  x  $\in$  carrier"
  by (auto simp: is_open_def carr_def)
```

lemma (in carrier) carrierE [elim]:

```
  "[[ x  $\in$  carrier;
     $\bigwedge$ t. [[ t open; x  $\in$  t ]  $\Longrightarrow$  R
  ]  $\Longrightarrow$  R"
  by (auto simp: is_open_def carr_def)
```

lemma (in carrier) subM:

```
  "[[ t  $\in$  M; M  $\subseteq$  T ]  $\Longrightarrow$  t open"
  by (auto simp: is_open_def)
```

lemma (in carrier) topeqI [intro!]:

```
  fixes S (structure)
  shows "[[  $\bigwedge$ m. m open1  $\Longrightarrow$  m open2;
     $\bigwedge$ m. m open2  $\Longrightarrow$  m open1 ]
     $\Longrightarrow$  T = S"
  by (auto simp: is_open_def)
```

locale topology = carrier T for T (structure) +

```
  assumes Int_open [intro!]: "[[ x open; y open ]  $\Longrightarrow$  x  $\cap$  y open"
  and union_open [intro]: " $\forall$ m  $\in$  M. m open  $\Longrightarrow$   $\bigcup$  M open"
```

lemma topologyI:

```
  "[[  $\bigwedge$  x y. [[ is_open T x; is_open T y ]  $\Longrightarrow$  is_open T (x  $\cap$  y);
     $\bigwedge$  M.  $\forall$  m  $\in$  M. is_open T m  $\Longrightarrow$  is_open T ( $\bigcup$  M)
  ]  $\Longrightarrow$  topology T"
  by (auto simp: topology_def)
```

lemma (in topology) Un_open [intro!]:

```
  assumes abopen: "A open" "B open"
  shows "A  $\cup$  B open"
```

proof-

```
  have " $\bigcup$ {A, B} open" using abopen
```

```

    by fast
  thus ?thesis by simp
qed

```

Common definitions of topological spaces require that the empty set and the carrier set of the space be open. With our definition, however, the carrier is implicitly given as the union of all open sets; therefore it is trivially open. The empty set is open by the laws of HOLs typed set theory.

```

lemma (in topology) empty_open [iff]: "{} open"
proof-
  have "⋃ {} open" by fast
  thus ?thesis by simp
qed

```

```

lemma (in topology) carrier_open [iff]: "carrier open"
  by (auto simp: carr_def intro: openI)

```

```

lemma (in topology) open_kriterion:
  assumes t_contains_open: "⋀ x. x ∈ t ⇒ ∃ t'. t' open ∧ x ∈ t' ∧ t' ⊆ t"
  shows "t open"
proof-
  let ?M = "⋃ x ∈ t. {t'. t' open ∧ x ∈ t' ∧ t' ⊆ t}"
  have "∀ m ∈ ?M. m open" by simp
  hence "⋃ ?M open" by auto
  moreover have "t = ⋃ ?M"
    by (auto dest!: t_contains_open)
  ultimately show ?thesis
    by simp
qed

```

We can obtain a topology from a set of basic open sets by closing the set under finite intersections and arbitrary unions.

```

inductive_set
  topo :: "'a set set ⇒ 'a top"
  for B :: "'a set set"
where
  basic [intro]: "x ∈ B ⇒ x ∈ topo B"
| inter [intro]: "[x ∈ topo B; y ∈ topo B] ⇒ x ∩ y ∈ topo B"
| union [intro]: "(⋀ x. x ∈ M ⇒ x ∈ topo B) ⇒ ⋃ M ∈ topo B"

locale topobase = carrier T for B and T (structure) +
  defines "T ≡ topo B"

```

```

lemma (in topobase) topo_open:
  "t open = (t ∈ topo B)"
  by (auto simp: T_def is_open_def)

```

```

lemma (in topobase)
  basic [intro]: "t ∈ B ⇒ t open" and
  inter [intro]: "[x open; y open] ⇒ (x ∩ y) open" and
  union [intro]: "(⋀ t. t ∈ M ⇒ t open) ⇒ ⋃ M open"
  by (auto simp: topo_open)

```

```

lemma (in topobase) topo_induct
  [case_names basic inter union, induct set: topo, consumes 1]:
  assumes opn: "x open"
  and bas: " $\bigwedge x. x \in B \implies P x$ "
  and int: " $\bigwedge x y. [x \text{ open}; P x; y \text{ open}; P y] \implies P (x \cap y)$ "
  and uni: " $\bigwedge M. (\forall t \in M. t \text{ open} \wedge P t) \implies P (\bigcup M)$ "
  shows "P x"
proof-
  from opn have "x  $\in$  topo B" by (simp add: topo_open)
  thus ?thesis
    by induct (auto intro: bas int intro!:uni simp: topo_open)
qed

lemma topo_topology [iff]:
  "topology (topo B)"
  by (auto intro!: union topologyI simp: is_open_def)

lemma topo_mono:
  assumes asubb: "A  $\subseteq$  B"
  shows "topo A  $\subseteq$  topo B"
proof
  fix m assume mintopoa: "m  $\in$  topo A"
  hence "A  $\subseteq$  B  $\longrightarrow$  m  $\in$  topo B"
    by (rule topo.induct) auto
  with asubb show "m  $\in$  topo B"
    by auto
qed

lemma topo_open_imp:
  fixes A and S (structure) defines "S  $\equiv$  topo A"
  fixes B and T (structure) defines "T  $\equiv$  topo B"
  shows "[[ A  $\subseteq$  B; x open1 ]  $\implies$  x open2" (is "PROP ?P")
proof -
  interpret A_S: topobase A S by fact
  interpret topobase B T by fact
  show "PROP ?P" by (auto dest: topo_mono iff: A_S.topo_open topo_open)
qed

lemma (in topobase) carrier_topo: "carrier =  $\bigcup B$ "
proof
  show "carrier  $\subseteq$   $\bigcup B$ "
  proof
    fix x assume "x  $\in$  carrier"
    then obtain t where "t open" and "x  $\in$  t" ..
    thus "x  $\in$   $\bigcup B$ " by (induct, auto)
  qed
qed (auto iff: topo_open)

Topological subspace

locale subtopology = carrier S + carrier T for S (structure) and T (structure) +
  assumes subtop[iff]: "s open = ( $\exists t. t \text{ open}_2 \wedge s = t \cap \text{carrier}$ )"

```

```

lemma subtopologyI:
  fixes S (structure)
  fixes T (structure)
  assumes H1: " $\bigwedge s. s \text{ open} \implies \exists t. t \text{ open}_2 \wedge s = t \cap \text{carrier}$ "
  and      H2: " $\bigwedge t. t \text{ open}_2 \implies t \cap \text{carrier} \text{ open}$ "
  shows "subtopology S T"
by (auto simp: subtopology_def intro: prems)

lemma (in subtopology) subtopologyE [elim]:
  assumes major: "s open"
  and      minor: " $\bigwedge t. [ t \text{ open}_2; s = t \cap \text{carrier} ] \implies R$ "
  shows "R"
using prems by auto

lemma (in subtopology) subtopI [intro]:
  "t open2  $\implies$  t  $\cap$  carrier open"
by auto

lemma (in subtopology) carrier_subset:
  "carrier1  $\subseteq$  carrier2"
by auto

lemma (in subtopology) subtop_sub:
  assumes "topology T"
  assumes carrSopen: "carrier1 open2"
  and s_open: "s open1"
  shows "s open2"
proof -
  interpret topology T by fact
  show ?thesis using prems by auto
qed

lemma (in subtopology) subtop_topology [iff]:
  assumes "topology T"
  shows "topology S"
proof -
  interpret topology T by fact
  show ?thesis proof (rule topologyI)
    fix u v assume uopen: "u open" and vopen: "v open"
    thus "u  $\cap$  v open" by (auto simp add: Int_ac)
  next
    fix M assume msub: " $\forall m \in M. m \text{ open}$ "
    let ?N = "{x. x open2  $\wedge$  x  $\cap$  carrier  $\in$  M}"
    have " $\bigcup ?N \text{ open}_2$ " by auto
    hence " $\bigcup ?N \cap \text{carrier} \text{ open}$ " ..
    moreover have " $\bigcup ?N \cap \text{carrier} = \bigcup M$ "
  proof
    show " $\bigcup M \subseteq \bigcup ?N \cap \text{carrier}$ "
  proof
    fix x assume "x  $\in$   $\bigcup M$ "
    then obtain s where sinM: "s  $\in$  M" and xins: "x  $\in$  s"
    by auto
  end
end

```

```

    from msub sinM have s_open: "s open" ..
  then obtain t
    where t_open: "t open2" and s_inter: "s = t ∩ carrier" by auto
  with xins have xint: "x∈t" and xpoint: "x ∈ carrier" by auto
  moreover
  from t_open s_inter sinM have "t ∈ ?N" by auto
  ultimately show "x ∈ ⋃?N ∩ carrier"
    by auto
  qed
  qed auto
  finally show "⋃M open" .
  qed
  qed

```

```

lemma subtop_lemma:
  fixes A and S (structure) defines "S ≡ topo A"
  fixes B and T (structure) defines "T ≡ topo B"
  assumes Asub: "A = (⋃t∈B. { t ∩ ⋃A })"
  shows "subtopology S T"
proof -
  interpret A_S: topobase A S by fact
  interpret topobase B T by fact
  show ?thesis proof (rule subtopologyI)
    fix s assume "s open1"
    thus "∃t. t open2 ∧ s = t ∩ carrier"
    proof induct
      case (basic s) with Asub
        obtain t where tB: "t ∈ B" and stA: "s = t ∩ ⋃A" by blast
        thus ?case by (auto simp: A_S.carrier_topo)
      next case (inter s t) thus ?case by auto
      next case (union M)
        let ?N = "⋃{u. u open2 ∧ (∃m∈M. m = u ∩ carrier)}"
        have "?N open2" and "⋃M = ?N ∩ carrier" using union by auto
        thus ?case by auto
    qed
  next
    fix t assume "t open2"
    thus "t ∩ carrier open"
    proof induct
      case (basic u) with Asub show ?case
        by (auto simp: A_S.carrier_topo)
      next case (inter u v)
        hence "(u ∩ carrier) ∩ (v ∩ carrier) open" by auto
        thus ?case by (simp add: Int_ac)
      next case (union M)
        let ?N = "⋃{s. ∃m∈M. s = m ∩ carrier}"
        from union have "?N open" and "?N = ⋃M ∩ carrier" by auto
        thus ?case by auto
    qed
  qed
  qed
  qed

```

Sample topologies

definition

```
trivial_top :: "'a top" where
"trivial_top = {{}}"
```

definition

```
discrete_top :: "'a set  $\Rightarrow$  'a set set" where
"discrete_top X = Pow X"
```

definition

```
indiscrete_top :: "'a set  $\Rightarrow$  'a set set" where
"indiscrete_top X = {{}, X}"
```

definition

```
order_base :: "('a::order) set  $\Rightarrow$  'a set set" where
"order_base A = ( $\bigcup_{x \in A} \{y. y \in A \wedge x \leq y\}$ )"
```

definition

```
order_top :: "('a::order) set  $\Rightarrow$  'a set set" where
"order_top X = topo(order_base X)"
```

```
locale trivial = carrier +
defines "T  $\equiv$  {{}}"
```

```
lemma (in trivial) open_iff [iff]:
"m open = (m = {})"
by (auto simp: T_def is_open_def)
```

```
lemma trivial_topology:
fixes T (structure) defines "T  $\equiv$  {{}}"
shows "topology T"
proof -
interpret trivial T by fact
show ?thesis by (auto intro: topologyI)
qed
```

```
lemma empty_carrier_implies_trivial:
fixes S (structure) assumes "topology S"
fixes T (structure) defines "T  $\equiv$  {{}}"
shows "carrier = {}  $\implies$  S = T" (is "PROP ?P")
proof -
interpret topology S by fact
interpret trivial T by fact
show "PROP ?P" by (auto intro!: topeqI)
qed
```

```
locale discrete = carrier T for X and T (structure) +
defines "T  $\equiv$  discrete_top X"
```

```
lemma (in discrete) carrier:
"carrier = X"
by (auto intro!: carrierI elim!: carrierE)
(auto simp: discrete_top_def T_def is_open_def)
```

```

lemma (in discrete) open_iff [iff]:
  "t open = (t ∈ Pow carrier)"
proof-
  have "t open = (t ∈ Pow X)"
    by (auto simp: T_def discrete_top_def is_open_def)
  thus ?thesis by (simp add: carrier)
qed

lemma discrete_topology: "topology (discrete_top X)"
  by (auto intro!: topologyI simp: is_open_def discrete_top_def)
  blast

locale indiscrete = carrier T for X and T (structure) +
  defines "T ≡ indiscrete_top X"

lemma (in indiscrete) carrier:
  "X = carrier"
  by (auto intro!: carrierI elim!: carrierE)
  (auto simp: T_def indiscrete_top_def is_open_def)

lemma (in indiscrete) open_iff [iff]:
  "t open = (t = {} ∨ t = carrier)"
proof-
  have "t open = (t = {} ∨ t = X)"
    by (auto simp: T_def indiscrete_top_def is_open_def)
  thus ?thesis by (simp add: carrier)
qed

lemma indiscrete_topology: "topology (indiscrete_top X)"
  by (rule topologyI) (auto simp: is_open_def indiscrete_top_def)

locale orderbase =
  fixes X and B
  defines "B ≡ order_base X"

locale ordertop1 = orderbase X B + topobase B T for X and B and T (structure)

locale ordertop = carrier T for X and T (structure) +
  defines "T ≡ order_top X"

lemma (in ordertop) ordertop_open:
  "t open = (t ∈ order_top X)"
  by (auto simp: T_def is_open_def)

lemma ordertop_topology [iff]:
  "topology (order_top X)"
  by (auto simp: order_top_def)

```

1.3 Neighbourhoods

definition

```

nhd :: "'a top ⇒ 'a ⇒ 'a set set" ( "nhds" ) where
  "nhd T x = {U. U ⊆ carr T ∧ (∃ m. is_open T m ∧ x ∈ m ∧ m ⊆ U)}"

```

```

lemma (in carrier) nhdI [intro]:
  "[ U ⊆ carrier; m open; x ∈ m; m ⊆ U ] ⇒ U ∈ nhds x"
  by (auto simp: nhd_def)

lemma (in carrier) nhdE [elim]:
  "[ U ∈ nhds x; ∧m. [ U ⊆ carrier; m open; x ∈ m; m ⊆ U ] ⇒ R ] ⇒ R"
  by (auto simp: nhd_def)

lemma (in carrier) elem_in_nhd:
  "U ∈ nhds x ⇒ x ∈ U"
  by auto

lemma (in carrier) carrier_nhd [intro]: "x ∈ carrier ⇒ carrier ∈ nhds x"
  by auto

lemma (in carrier) empty_not_nhd [iff]:
  "{} ∉ nhds x"
  by auto

lemma (in carrier) nhds_greater:
  "[V ⊆ carrier; U ⊆ V; U ∈ nhds x] ⇒ V ∈ nhds x"
  by (erule nhdE) blast

lemma (in topology) nhds_inter:
  assumes nhdU: "U ∈ nhds x"
  and      nhdV: "V ∈ nhds x"
  shows "(U ∩ V) ∈ nhds x"
proof-
  from nhdU obtain u where
    Usub: "U ⊆ carrier" and
    uT:   "u open" and
    xu:   "x ∈ u" and
    usub: "u ⊆ U"
  by auto
  from nhdV obtain v where
    Vsub: "V ⊆ carrier" and
    vT:   "v open" and
    xv:   "x ∈ v" and
    vsub: "v ⊆ V"
  by auto
  from Usub Vsub have "U ∩ V ⊆ carrier" by auto
  moreover from uT vT have "u ∩ v open" ..
  moreover from xu xv have "x ∈ u ∩ v" ..
  moreover from usub vsub have "u ∩ v ⊆ U ∩ V" by auto
  ultimately show ?thesis by auto
qed

lemma (in carrier) sub_nhd:
  "U ∈ nhds x ⇒ ∃V ∈ nhds x. V ⊆ U ∧ (∀ z ∈ V. U ∈ nhds z)"
  by (auto elim!: nhdE intro: nhdI)

lemma (in ordertop1) l1:

```

```

    assumes mopen: "m open"
    and xpoint: "x ∈ X"
    and ypoint: "y ∈ X"
    and xley: "x ≤ y"
    and xinm: "x ∈ m"
    shows "y ∈ m"
    using mopen xinm
proof induct
  case (basic U) thus ?case
    by (auto simp: B_def order_base_def ypoint
        intro: xley dest: order_trans)
qed auto

lemma (in ordertop1)
  assumes xpoint: "x ∈ X" and ypoint: "y ∈ X" and xley: "x ≤ y"
  shows "nhds x ⊆ nhds y"
proof
  fix u assume "u ∈ nhds x"
  then obtain m where "m open"
    and "m ⊆ u" and "u ⊆ carrier" and "x ∈ m"
    by auto
  with xpoint ypoint xley
  show "u ∈ nhds y"
    by (auto dest: l1)
qed

```

1.4 Closed sets

A set is closed if its complement is open.

definition

```

is_closed :: "'a top ⇒ 'a set ⇒ bool" ("_ closed" [50] 50) where
  "is_closed T s ⟷ is_open T (carr T - s)"

```

```

lemma (in carrier) closedI:
  "(carrier - s) open ⟹ s closed"
  by (auto simp: is_closed_def)

```

```

lemma (in carrier) closedE:
  "[[ s closed; (carrier - s) open ⟹ R ] ⟹ R"
  by (auto simp: is_closed_def)

```

```

lemma (in topology) empty_closed [iff]:
  "{} closed"
  by (auto intro!: closedI)

```

```

lemma (in topology) carrier_closed [iff]:
  "carrier closed"
  by (auto intro!: closedI)

```

```

lemma (in carrier) compl_open_closed:
  assumes mopen: "m open"
  shows "(carrier - m) closed"

```

```

proof (rule closedI)
  from mopen have "m  $\subseteq$  carrier"
    by auto
  hence "carrier - (carrier - m) = m"
    by (simp add: double_diff)
  thus "carrier - (carrier - m) open"
    using mopen by simp
qed

lemma (in carrier) compl_open_closed1:
  "[[ m  $\subseteq$  carrier; (carrier - m) closed ]  $\implies$  m open"
  by (auto elim: closedE simp: diffsimps)

lemma (in carrier) compl_closed_iff [iff]:
  "m  $\subseteq$  carrier  $\implies$  (carrier - m) closed = (m open)"
  by (auto dest: compl_open_closed1 intro: compl_open_closed)

lemma (in topology) Un_closed [intro!]:
  "[[ x closed; y closed ]  $\implies$  x  $\cup$  y closed"
  by (auto simp: Diff_Un elim!: closedE intro!: closedI)

lemma (in topology) inter_closed:
  assumes xsclosed: " $\bigwedge x. x \in S \implies x$  closed"
  shows " $\bigcap S$  closed"
proof (rule closedI)
  let ?M = "{m.  $\exists x \in S. m = \text{carrier} - x$ }"
  have " $\forall m \in ?M. m$  open"
    by (auto dest: xsclosed elim: closedE)
  hence " $\bigcup ?M$  open" ..
  moreover have " $\bigcup ?M = \text{carrier} - \bigcap S$ " by auto
  ultimately show "carrier -  $\bigcap S$  open" by auto
qed

corollary (in topology) Int_closed [intro!]:
  assumes abclosed: "A closed" "B closed"
  shows "A  $\cap$  B closed"
proof-
  from prems have " $\bigcap \{A, B\}$  closed"
    by (blast intro!: inter_closed)
  thus ?thesis by simp
qed

lemma (in topology) closed_diff_open:
  assumes aclosed: "A closed"
  and bopen: "B open"
  shows "A - B closed"
proof (rule closedI)
  from aclosed have "carrier - A open"
    by (rule closedE)
  moreover from bopen have "carrier  $\cap$  B open" by auto
  ultimately have "(carrier - A)  $\cup$  (carrier  $\cap$  B) open" ..
  thus "carrier - (A - B) open" by (simp add: diff_diff)
qed

```

```

lemma (in topology) open_diff_closed:
assumes aclosed: "A closed"
    and bopen: "B open"
    shows "B - A open"
proof-
  from aclosed have "carrier - A open"
    by (rule closedE)
  hence "(carrier - A) ∩ B open" using bopen ..
  moreover from bopen have "B ⊆ carrier"
    by auto
  hence "(carrier - A) ∩ B = B - A" by auto
  ultimately show "B - A open" by simp
qed

```

1.5 Core, closure, and frontier of a set

definition

```

cor :: "'a top ⇒ 'a set ⇒ 'a set"          ("corez") where
"cor T s = (⋃{m. is_open T m ∧ m ⊆ s})"

```

definition

```

clsr :: "'a top ⇒ 'a set ⇒ 'a set"          ("closurez") where
"clsr T a = (⋂{c. is_closed T c ∧ a ⊆ c})"

```

definition

```

frt :: "'a top ⇒ 'a set ⇒ 'a set"          ("frontierz") where
"frt T s = clsr T s - cor T s"

```

1.5.1 Core

lemma (in carrier) coreI:

```

"[[m open; m ⊆ s; x ∈ m]] ⇒ x ∈ core s"
by (auto simp: cor_def)

```

lemma (in carrier) coreE:

```

"[[x ∈ core s; ∧m. [m open; m ⊆ s; x ∈ m]] ⇒ R]] ⇒ R"
by (auto simp: cor_def)

```

lemma (in topology) core_open [iff]:

```

"core a open"
by (auto simp: cor_def)

```

lemma (in carrier) core_subset:

```

"core a ⊆ a"
by (auto simp: cor_def)

```

lemmas (in carrier) core_subsetD = subsetD [OF core_subset, standard]

lemma (in carrier) core_greatest:

```

"[[m open; m ⊆ a]] ⇒ m ⊆ core a"
by (auto simp: cor_def)

```

```

lemma (in carrier) core_idem [simp]:
  "core (core a) = core a"
  by (auto simp: cor_def)

lemma (in carrier) open_core_eq [simp]:
  "a open  $\implies$  core a = a"
  by (auto simp: cor_def)

lemma (in topology) core_eq_open:
  "core a = a  $\implies$  a open"
  by (auto elim: subst)

lemma (in topology) core_iff:
  "a open = (core a = a)"
  by (auto intro: core_eq_open)

lemma (in carrier) core_mono:
  "a  $\subseteq$  b  $\implies$  core a  $\subseteq$  core b"
  by (auto simp: cor_def)

lemma (in topology) core_Int [simp]:
  "core (a  $\cap$  b) = core a  $\cap$  core b"
  by (auto simp: cor_def)

lemma (in carrier) core_nhds:
  "[[ U  $\subseteq$  carrier; x  $\in$  core U ]  $\implies$  U  $\in$  nhds x"
  by (auto elim!: coreE)

lemma (in carrier) nhds_core:
  "U  $\in$  nhds x  $\implies$  x  $\in$  core U"
  by (auto intro: coreI)

lemma (in carrier) core_nhds_iff:
  "U  $\subseteq$  carrier  $\implies$  (x  $\in$  core U) = (U  $\in$  nhds x)"
  by (auto intro: core_nhds nhds_core)

```

1.5.2 Closure

```

lemma (in carrier) closureI [intro]:
  "( $\bigwedge$ c. [[c closed; a  $\subseteq$  c]  $\implies$  x  $\in$  c]  $\implies$  x  $\in$  closure a"
  by (auto simp: clsr_def)

lemma (in carrier) closureE [elim]:
  "[[ x  $\in$  closure a;  $\neg$  c closed  $\implies$  R;  $\neg$  a  $\subseteq$  c  $\implies$  R; x  $\in$  c  $\implies$  R ]  $\implies$  R"
  by (auto simp: clsr_def)

lemma (in carrier) closure_least:
  "s closed  $\implies$  closure s  $\subseteq$  s"
  by auto

lemma (in carrier) subset_closure:
  "s  $\subseteq$  closure s"
  by auto

```

```

lemma (in topology) closure_carrier [simp]:
  "closure carrier = carrier"
  by auto

lemma (in topology) closure_subset:
  "A  $\subseteq$  carrier  $\implies$  closure A  $\subseteq$  carrier"
  by auto

lemma (in topology) closure_closed [iff]:
  "closure a closed"
  by (auto simp: clsr_def intro: inter_closed)

lemma (in carrier) closure_idem [simp]:
  "closure (closure s) = closure s"
  by (auto simp: clsr_def)

lemma (in carrier) closed_closure_eq [simp]:
  "a closed  $\implies$  closure a = a"
  by (auto simp: clsr_def)

lemma (in topology) closure_eq_closed:
  "closure a = a  $\implies$  a closed"
  by (erule subst) simp

lemma (in topology) closure_iff:
  "a closed = (closure a = a)"
  by (auto intro: closure_eq_closed)

lemma (in carrier) closure_mono1:
  "mono (closure)"
  by (rule, auto simp: clsr_def)

lemma (in carrier) closure_mono:
  "a  $\subseteq$  b  $\implies$  closure a  $\subseteq$  closure b"
  by (auto simp: clsr_def)

lemma (in topology) closure_Un [simp]:
  "closure (a  $\cup$  b) = closure a  $\cup$  closure b"
  by (rule, blast) (auto simp: clsr_def)

```

1.5.3 Frontier

```

lemma (in carrier) frontierI:
  "[[x  $\in$  closure s; x  $\in$  core s  $\implies$  False]]  $\implies$  x  $\in$  frontier s"
  by (auto simp: frt_def)

lemma (in carrier) frontierE:
  "[[ x  $\in$  frontier s; [[x  $\in$  closure s; x  $\in$  core s  $\implies$  False]]  $\implies$  R ]  $\implies$  R"
  by (auto simp: frt_def)

lemma (in topology) frontier_closed [iff]:

```

```

"frontier s closed"
by (unfold frt_def)
(intro closure_closed core_open closed_diff_open)

lemma (in carrier) frontier_Un_core:
  "frontier s  $\cup$  core s = closure s"
  by (auto dest: subsetD [OF core_subset] simp: frt_def)

lemma (in carrier) frontier_Int_core:
  "frontier s  $\cap$  core s = {}"
  by (auto simp: frt_def)

lemma (in topology) closure_frontier [simp]:
  "closure (frontier a) = frontier a"
  by simp

lemma (in topology) frontier_carrier [simp]:
  "frontier carrier = {}"
  by (auto simp: frt_def)

```

Hence frontier is not monotone. Also $\text{core}_T (\text{frontier}_T A) = \{\}$ is not a theorem as illustrated by the following counter example. By the way: could the counter example be proved using an instantiation?

```

lemma counter_example_core_frontier:
  fixes X defines [simp]: "X  $\equiv$  (UNIV::nat set)"
  fixes T (structure) defines "T  $\equiv$  indiscrete_top X"
  shows "core (frontier {0}) = X"
proof -
  interpret indiscrete X T by fact
  have "core {0} = {}"
    by (auto simp add: carrier [symmetric] cor_def)
  moreover have "closure {0} = UNIV"
    by (auto simp: clsr_def carrier [symmetric] is_closed_def)
  ultimately have "frontier {0} = UNIV"
    by (auto simp: frt_def)
  thus ?thesis
    by (auto simp add: cor_def carrier [symmetric])
qed

```

1.5.4 Adherent points

definition

```

adhs :: "'a top  $\Rightarrow$  'a  $\Rightarrow$  'a set  $\Rightarrow$  bool"      (infix "adhz" 50) where
"adhs T x A  $\longleftrightarrow$  ( $\forall U \in \text{nhd } T x. U \cap A \neq \{\}$ )"

```

```

lemma (in carrier) adhCE [elim?]:
"[x adh A; U  $\notin$  nhds x  $\implies$  R; U  $\cap$  A  $\neq$  {}  $\implies$  R]  $\implies$  R"
  by (unfold adhs_def) auto

```

```

lemma (in carrier) adhI [intro]:
"( $\bigwedge U. U \in \text{nhds } x \implies U \cap A \neq \{\}$ )  $\implies$  x adh A"
  by (unfold adhs_def) simp

```

```

lemma (in carrier) closure_imp_adh:
  assumes asub: "A  $\subseteq$  carrier"
  and closure: "x  $\in$  closure A"
  shows "x adh A"
proof
  fix U assume unhd: "U  $\in$  nhds x"
  show "U  $\cap$  A  $\neq$  {}"
  proof
    assume UA: "U  $\cap$  A = {}"
    from unhd obtain V where "V open" "x  $\in$  V" and VU: "V  $\subseteq$  U" ..
    moreover from UA VU have "V  $\cap$  A = {}" by auto
    ultimately show "False" using asub closure
      by (auto dest!: compl_open_closed simp: clsr_def)
  qed
qed

```

```

lemma (in carrier) adh_imp_closure:
  assumes xpoint: "x  $\in$  carrier"
  and adh: "x adh A"
  shows "x  $\in$  closure A"
proof (rule ccontr)
  assume notclosure: "x  $\notin$  closure A"
  then obtain C
    where closed: "C closed"
    and asubc: "A  $\subseteq$  C"
    and xnotinc: "x  $\notin$  C"
    by (auto simp: clsr_def)
  from closed have "carrier - C open" by (rule closedE)
  moreover from xpoint xnotinc have "x  $\in$  carrier - C" by simp
  ultimately have "carrier - C  $\in$  nhds x" by auto
  with adh have "(carrier - C)  $\cap$  A  $\neq$  {}"
    by (auto elim: adhCE)
  with asubc show "False" by auto
qed

```

```

lemma (in topology) closed_adh:
  assumes Asub: "A  $\subseteq$  carrier"
  shows "A closed = ( $\forall$  x  $\in$  carrier. x adh A  $\longrightarrow$  x  $\in$  A)"
proof
  assume "A closed"
  hence AA: "closure A = A"
    by auto
  thus "( $\forall$  x  $\in$  carrier. x adh A  $\longrightarrow$  x  $\in$  A)"
    by (fast dest!: adh_imp_closure)
next assume adhA: " $\forall$  x  $\in$  carrier. x adh A  $\longrightarrow$  x  $\in$  A"
  have "closure A  $\subseteq$  A"
  proof
    fix x assume xclosure: "x  $\in$  closure A"
    hence "x  $\in$  carrier" using Asub by (auto dest: closure_subset)
    with xclosure show "x  $\in$  A" using Asub adhA
      by (auto dest!: closure_imp_adh)
  qed
  thus "A closed" by (auto intro: closure_eq_closed)

```

qed

```
lemma (in carrier) adh_closure_iff:
  "[[ A ⊆ carrier; x ∈ carrier ]] ⇒ (x adh A) = (x ∈ closure A)"
  by (auto dest: adh_imp_closure closure_imp_adh)
```

1.6 More about closure and core

```
lemma (in topology) closure_complement [simp]:
  shows "closure (carrier - A) = carrier - core A"
proof
  have "closure (carrier - A) ⊆ carrier"
    by (auto intro: closure_subset)
  moreover have "closure (carrier - A) ∩ core A = {}"
proof (rule seteqI, clarsimp)
  fix x assume xclosure: "x ∈ closure (carrier - A)"
  hence xadh: "x adh carrier - A"
    by (auto intro: closure_imp_adh)
  moreover assume xcore: "x ∈ core A"
  hence "core A ∈ nhds x"
    by auto
  ultimately have "core A ∩ (carrier - A) ≠ {}"
    by (auto elim: adhCE)
  thus "False" by (auto dest: core_subsetD)
qed auto
ultimately show "closure (carrier - A) ⊆ carrier - core A"
  by auto
next
  show "carrier - core A ⊆ closure (carrier - A)"
    by (auto simp: cor_def clsr_def is_closed_def)
qed
```

```
lemma (in carrier) core_complement [simp]:
  assumes asub: "A ⊆ carrier"
  shows "core (carrier - A) = carrier - closure A"
proof
  show "carrier - closure A ⊆ core (carrier - A)"
    by (auto simp: cor_def clsr_def is_closed_def)
next
  have "core (carrier - A) ⊆ carrier"
    by (auto elim!: coreE)
  moreover have "core (carrier - A) ∩ closure A = {}"
proof auto
  fix x assume "x ∈ core (carrier - A)"
  hence "(carrier - A) ∈ nhds x"
    by (auto iff: core_nhds_iff)
  moreover assume "x ∈ closure A"
  ultimately have "A ∩ (carrier - A) ≠ {}" using asub
    by (auto dest!: closure_imp_adh elim!: adhCE)
  thus "False" by auto
qed
ultimately show "core (carrier - A) ⊆ carrier - closure A"
```

by auto
qed

```
lemma (in carrier) core_closure_diff_empty [simp]:
  assumes asub: "A ⊆ carrier"
  shows "core (closure A - A) = {}"
proof auto
  fix x assume "x ∈ core (closure A - A)"
  then obtain m where
    mopen: "m open" and
    xinm: "x ∈ m" and
    msub: "m ⊆ closure A" and
    minter: "m ∩ A = {}"
  by (auto elim!: coreE)
  from xinm msub have "x adh A" using asub
  by (auto dest: closure_imp_adh)
  moreover from xinm mopen have "m ∈ nhds x"
  by auto
  ultimately have "m ∩ A ≠ {}" by (auto elim: adhCE)
  with minter show "False" by auto
qed
```

1.7 Dense sets

definition

```
is_densein :: "'a top ⇒ 'a set ⇒ 'a set ⇒ bool" (infix "densein" 50) where
  "is_densein T A B ⟷ B ⊆ clsr T A"
```

definition

```
is_dense :: "'a top ⇒ 'a set ⇒ bool" ("_ dense" [50] 50) where
  "is_dense T A = is_densein T A (carr T)"
```

```
lemma (in carrier) densinI [intro!]: "B ⊆ closure A ⟹ A densein B"
  by (auto simp: is_densein_def)
```

```
lemma (in carrier) denseinE [elim!]: "[ A densein B; B ⊆ closure A ⟹ R ] ⟹ R"
  by (auto simp: is_densein_def)
```

```
lemma (in carrier) denseI [intro!]: "carrier ⊆ closure A ⟹ A dense"
  by (auto simp: is_dense_def)
```

```
lemma (in carrier) denseE [elim]: "[ A dense; carrier ⊆ closure A ⟹ R ] ⟹ R"
  by (auto simp: is_dense_def)
```

```
lemma (in topology) dense_closure_eq [dest]:
  "[ A dense; A ⊆ carrier ] ⟹ closure A = carrier"
  by (auto dest: closure_subset)
```

```
lemma (in topology) dense_lemma:
  "A ⊆ carrier ⟹ carrier - (closure A - A) dense"
  by auto
```

```

lemma (in topology) ex_dense_closure_inter:
  assumes ssub: "S  $\subseteq$  carrier"
  shows " $\exists D C. D \text{ dense} \wedge C \text{ closed} \wedge S = D \cap C$ "

```

proof-

```

  let ?D = "carrier - (closure S - S)" and
      ?C = "closure S"
  from ssub have "?D dense" by auto
  moreover have "?C closed" ..
  moreover from ssub
  have "(carrier - (closure S - S))  $\cap$  closure S = S"
    by (simp add: diff_diff_inter subset_closure)
  ultimately show ?thesis
    by auto

```

qed

```

lemma (in topology) ex_dense_closure_interE:
  assumes ssub: "S  $\subseteq$  carrier"
  and H: " $\bigwedge D C. [ D \subseteq \text{carrier}; C \subseteq \text{carrier}; D \text{ dense}; C \text{ closed}; S = D \cap C ] \implies R$ "
  shows "R"

```

proof-

```

  let ?D = "(carrier - (closure S - S))"
  and ?C = "closure S"
  have "?D  $\subseteq$  carrier" by auto
  moreover from prems have "?C  $\subseteq$  carrier"
    by (auto dest!: closure_subset)
  moreover from prems have "?D dense" by auto
  moreover have "?C closed" ..
  moreover from ssub have "S = ?D  $\cap$  ?C"
    by (simp add: diff_diff_inter subset_closure)
  ultimately show ?thesis
    by (rule H)

```

qed

1.8 Continuous functions

definition

```

INJ :: "'a set  $\implies$  'b set  $\implies$  ('a  $\implies$  'b) set" where
"INJ A B = {f. f : A  $\rightarrow$  B  $\wedge$  inj_on f A}"

```

definition

```

SUR :: "'a set  $\implies$  'b set  $\implies$  ('a  $\implies$  'b) set" where
"SUR A B = {f. f : A  $\rightarrow$  B  $\wedge$  ( $\forall y \in B. \exists x \in A. y = f x$ )}"

```

definition

```

BIJ :: "'a set  $\implies$  'b set  $\implies$  ('a  $\implies$  'b) set" where
"BIJ A B = INJ A B  $\cap$  SUR A B"

```

definition

```

cnt :: "'a top  $\implies$  'b top  $\implies$  ('a  $\implies$  'b) set" where
"cnt S T = {f. f : carr S  $\rightarrow$  carr T  $\wedge$ 
  ( $\forall m. \text{is\_open } T m \implies \text{is\_open } S (\text{carr } S \cap (f \text{ -' } m))$ )}"

```

definition

```
HOM :: "'a top  $\Rightarrow$  'b top  $\Rightarrow$  ('a  $\Rightarrow$  'b) set" where
"HOM S T = {f. f  $\in$  cnt S T  $\wedge$  inv f  $\in$  cnt T S  $\wedge$  f  $\in$  BIJ (carr S) (carr T)}"
```

definition

```
homeo :: "'a top  $\Rightarrow$  'b top  $\Rightarrow$  bool" where
"homeo S T  $\longleftrightarrow$  ( $\exists$ h  $\in$  BIJ (carr S) (carr T). h  $\in$  cnt S T  $\wedge$  inv h  $\in$  cnt T S)"
```

definition

```
fimg :: "'b top  $\Rightarrow$  ('a  $\Rightarrow$  'b)  $\Rightarrow$  'a set set  $\Rightarrow$  'b set set" where
"fimg T f F = {v. v  $\subseteq$  carr T  $\wedge$  ( $\exists$  u  $\in$  F. f'u  $\subseteq$  v)}"
```

lemma domain_subset_vimage:

```
"f : A  $\rightarrow$  B  $\implies$  A  $\subseteq$  f-'B"
by (auto intro: funcset_mem)
```

lemma domain_inter_vimage:

```
"f : A  $\rightarrow$  B  $\implies$  A  $\cap$  f-'B = A"
by (auto intro: funcset_mem)
```

lemma funcset_vimage_diff:

```
"f : A  $\rightarrow$  B  $\implies$  A - f-'(B - C) = A  $\cap$  f-'C"
by (auto intro: funcset_mem)
```

locale func = S: carrier S + T: carrier T

```
for f and S (structure) and T (structure) and fimage +
assumes func [iff]: "f : carrier1  $\rightarrow$  carrier2"
defines "fimage  $\equiv$  fimg T f"
notes func_mem [simp, intro] = funcset_mem [OF func]
and domain_subset_vimage [iff] = domain_subset_vimage [OF func]
and domain_inter_vimage [simp] = domain_inter_vimage [OF func]
and vimage_diff [simp] = funcset_vimage_diff [OF func]
```

lemma (in func) fimageI [intro!]:

```
shows "[ v  $\subseteq$  carrier2; u  $\in$  F; f'u  $\subseteq$  v ]  $\implies$  v  $\in$  fimage F"
by (auto simp: fimg_def fimage_def)
```

lemma (in func) fimageE [elim!]:

```
"[ v  $\in$  fimage F;  $\bigwedge$ u. [ v  $\subseteq$  carrier2 ; u  $\in$  F; f'u  $\subseteq$  v ]  $\implies$  R ]  $\implies$  R"
by (auto simp: fimage_def fimg_def)
```

lemma cntI:

```
"[ f : carr S  $\rightarrow$  carr T;
  ( $\bigwedge$ m. is_open T m  $\implies$  is_open S (carr S  $\cap$  (f -' m))) ]
 $\implies$  f  $\in$  cnt S T"
by (auto simp: cnt_def)
```

lemma cntE:

```
"[ f  $\in$  cnt S T;
  [ f : carr S  $\rightarrow$  carr T;
     $\forall$ m. is_open T m  $\longrightarrow$  is_open S (carr S  $\cap$  (f -' m)) ]  $\implies$  P
]  $\implies$  P"
```

```

by (auto simp: cnt_def)

lemma cntCE:
  "[[ f ∈ cnt S T;
    [[ ¬ is_open T m; f : carr S → carr T ]] ⇒ P;
    [[ is_open S (carr S ∩ (f -' m)); f : carr S → carr T ]] ⇒ P
  ]] ⇒ P"
by (auto simp: cnt_def)

lemma cnt_fun:
  "f ∈ cnt S T ⇒ f : carr S → carr T"
by (auto simp add: cnt_def)

lemma cntD1:
  "[[ f ∈ cnt S T; x ∈ carr S ]] ⇒ f x ∈ carr T"
by (auto simp add: cnt_def intro: funcset_mem)

lemma cntD2:
  "[[ f ∈ cnt S T; is_open T m ]] ⇒ is_open S (carr S ∩ (f -' m))"
by (auto simp: cnt_def)

locale continuous = func +
  assumes continuous [dest, simp]:
  "m open₂ ⇒ carrier ∩ (f -' m) open"

lemma continuousI:
  fixes S (structure)
  fixes T (structure)
  assumes "f : carrier₁ → carrier₂"
  "∧m. m open₂ ⇒ carrier ∩ (f -' m) open"
  shows "continuous f S T"
using assms by (auto simp: continuous_def func_def continuous_axioms_def)

lemma continuousE:
  fixes S (structure)
  fixes T (structure)
  shows
  "[[ continuous f S T;
    [[ f : carrier₁ → carrier₂;
      ∀m. m open₂ ⇒ carrier₁ ∩ (f -' m) open ]] ⇒ P
  ]] ⇒ P"
by (auto simp: continuous_def func_def continuous_axioms_def)

lemma continuousCE:
  fixes S (structure)
  fixes T (structure)
  shows
  "[[ continuous f S T;
    [[ ¬ m open₂; f : carrier₁ → carrier₂ ]] ⇒ P;
    [[ carrier₁ ∩ (f -' m) open₁; f : carrier₁ → carrier₂ ]] ⇒ P
  ]] ⇒ P"
by (auto simp: continuous_def func_def continuous_axioms_def)

```

```

lemma (in continuous) closed_vimage [intro, simp]:
  assumes csubset: "c  $\subseteq$  carrier2 "
  and cclosed: "c closed2"
  shows "f -' c closed"
proof-
  from cclosed have "carrier2 - c open2" by (rule closedE)
  hence "carrier  $\cap$  f -' (carrier2 - c) open" by auto
  hence "carrier - f -' c open" by (auto simp: diffsimps)
  thus "f -' c closed" by (rule S.closedI)
qed

lemma continuousI2:
  fixes S (structure)
  fixes T (structure)
  assumes func: "f : carrier1  $\rightarrow$  carrier2"
  assumes R: " $\bigwedge$ c. [ c  $\subseteq$  carrier2; c closed2 ]  $\implies$  f -' c closed"
  shows "continuous f S T"
proof (rule continuous.intro)
  from func show "func f S T" by (auto simp: func_def)
next
  interpret S: carrier S .
  interpret T: carrier T .
  show "continuous_axioms f S T"
proof (rule continuous_axioms.intro)
  fix m let ?c = "carrier2 - m" assume "m open2"
  hence csubset: "?c  $\subseteq$  carrier2" and cclosed: "?c closed2"
  by auto
  hence "f -' ?c closed" by (rule R)
  hence "carrier - f -' ?c open"
  by (rule S.closedE)
  thus "carrier  $\cap$  f -' m open" by (simp add: funcset_vimage_diff [OF func])
qed
qed

lemma cnt_compose:
  "[[ f  $\in$  cnt S T; g  $\in$  cnt T U ]  $\implies$  (g  $\circ$  f)  $\in$  cnt S U "
  by (auto intro!: cntI funcset_comp elim!: cntE simp add: vimage_compose)

lemma continuous_compose:
  "[[ continuous f S T; continuous g T U ]  $\implies$  continuous (g  $\circ$  f) S U"
  by (auto intro!: continuousI funcset_comp
      elim!: continuousE simp add: vimage_compose)

lemma id_continuous:
  fixes T (structure)
  shows "continuous id T T"
proof(rule continuousI)
  show "id  $\in$  carrier  $\rightarrow$  carrier"
  by (auto intro: funcsetI)
next

```

```

interpret T: carrier T .
fix m assume mopen: "m open"
hence "m  $\subseteq$  carrier" by auto
hence "carrier  $\cap$  m = m" by auto
thus "carr T  $\cap$  id -' m open" using mopen
  by auto
qed

```

```

lemma (in discrete) continuous:
  fixes f and S (structure) and fimage
  assumes "func f T S" defines "fimage  $\equiv$  fimg S f"
  shows "continuous f T S"
proof -
  interpret func f T S fimage by fact fact
  show ?thesis by (auto intro!: continuousI)
qed

```

```

lemma (in indiscrete) continuous:
  fixes S (structure)
  assumes "topology S"
  fixes f and fimage
  assumes "func f S T" defines "fimage  $\equiv$  fimg T f"
  shows "continuous f S T"
proof -
  interpret S: topology S by fact
  interpret func f S T fimage by fact fact
  show ?thesis by (auto del: S.Int_open intro!: continuousI)
qed

```

1.9 Filters

```

definition
  fbas :: "'a top  $\Rightarrow$  'a set set  $\Rightarrow$  bool" ("fbasz") where
  "fbas T B  $\longleftrightarrow$   $\{\} \notin B \wedge B \neq \{\} \wedge$ 
    ( $\forall a \in B. \forall b \in B. \exists c \in B. c \subseteq a \cap b$ )"

```

```

definition
  filters :: "'a top  $\Rightarrow$  'a set set set" ("Filtersz") where
  "filters T = { F.  $\{\} \notin F \wedge \bigcup F \subseteq \text{carr T} \wedge$ 
    ( $\forall A B. A \in F \wedge B \in F \longrightarrow A \cap B \in F$ )  $\wedge$ 
    ( $\forall A B. A \in F \wedge A \subseteq B \wedge B \subseteq \text{carr T} \longrightarrow B \in F$ ) }"

```

```

definition
  ultr :: "'a top  $\Rightarrow$  'a set set  $\Rightarrow$  bool" ("ultraz") where
  "ultr T F  $\longleftrightarrow$  ( $\forall A. A \subseteq \text{carr T} \longrightarrow A \in F \vee (\text{carr T} - A) \in F$ )"

```

```

lemma filtersI [intro]:
  fixes T (structure)
  assumes a1: " $\{\} \notin F$ "
  and a2: " $\bigcup F \subseteq \text{carrier}$ "
  and a3: " $\bigwedge A B. [ A \in F; B \in F ] \Longrightarrow A \cap B \in F$ "
  and a4: " $\bigwedge A B. [ A \in F; A \subseteq B; B \subseteq \text{carrier} ] \Longrightarrow B \in F$ "
  shows "F  $\in$  Filters"

```

```

using a1 a2
by (auto simp add: filters_def intro: a3 a4)

lemma filtersE:
  assumes a1: "F ∈ filters T"
  and R: "[[ {} ∉ F;
            ⋃ F ⊆ carr T;
            ∀ A B. A ∈ F ∧ B ∈ F ⟶ A ∩ B ∈ F;
            ∀ A B. A ∈ F ∧ A ⊆ B ∧ B ⊆ carr T ⟶ B ∈ F
          ] ] ⟹ R"
  shows "R"
  using a1
  apply (simp add: filters_def)
  apply (rule R)
  apply ((erule conjE)+, assumption)+
  done

lemma filtersD1:
  "F ∈ filters T ⟹ {} ∉ F"
  by (erule filtersE)

lemma filtersD2:
  "F ∈ filters T ⟹ ⋃ F ⊆ carr T"
  by (erule filtersE)

lemma filtersD3:
  "[[ F ∈ filters T; A ∈ F; B ∈ F ] ] ⟹ A ∩ B ∈ F"
  by (blast elim: filtersE)

lemma filtersD4:
  "[[ F ∈ filters T; A ⊆ B; B ⊆ carr T; A ∈ F ] ] ⟹ B ∈ F"
  by (blast elim: filtersE)

locale filter = carrier T for F and T (structure) +
  assumes F_filter: "F ∈ Filters"
  notes not_empty [iff]      = filtersD1 [OF F_filter]
  and union_carr [iff]      = filtersD2 [OF F_filter]
  and filter_inter [intro!, simp] = filtersD3 [OF F_filter]
  and filter_greater [dest] = filtersD4 [OF F_filter]

lemma (in filter) elem_carrier [elim]:
  assumes A: "A ∈ F"
  assumes R: "[[ A ⊆ carrier; A ≠ {} ] ] ⟹ R"
  shows "R"
proof-
  have "⋃ F ⊆ carrier" ..
  thus ?thesis using A by (blast intro: R)
qed

```

```

lemma empty_filter [iff]: "{} ∈ filters T"
  by auto

lemma (in filter) contains_carrier [intro, simp]:
  assumes F_not_empty: "F ≠ {}"
  shows "carrier ∈ F"
proof-
  from F_not_empty obtain A where "A ⊆ carrier" "A ∈ F"
    by auto
  thus ?thesis by auto
qed

lemma nonempty_filter_implies_nonempty_carrier:
  fixes T (structure)
  assumes F_filter: "F ∈ Filters"
  and F_not_empty: "F ≠ {}"
  shows "carrier ≠ {}"
proof-
  from assms have "carrier ∈ F"
    by (auto dest!: filter.contains_carrier [OF filter.intro])
  thus ?thesis using F_filter
    by(auto dest: filtersD1)
qed

lemma carrier_singleton_filter:
  fixes T (structure)
  shows "carrier ≠ {} ⇒ {carrier} ∈ Filters"
  by auto

lemma (in topology) nhds_filter:
  "nhds x ∈ Filters"
  by (auto dest: nhds_greater intro!: filtersI nhds_inter)

lemma fimage_filter:
  fixes f and S (structure) and T (structure) and fimage
  assumes "func f S T" defines "fimage ≡ fimg T f"
  fixes F assumes "filter F S"
  shows "fimage F ∈ Filters2"
proof -
  interpret func f S T fimage by fact fact
  interpret filter F S by fact
  show ?thesis proof
    fix A B assume "A ∈ fimage F" "B ∈ fimage F"
    then obtain a b where
      AY: "A ⊆ carrier2" and aF: "a ∈ F" and fa: "f ` a ⊆ A" and
      BY: "B ⊆ carrier2" and bF: "b ∈ F" and fb: "f ` b ⊆ B"
    by (auto)
    from AY BY have "A ∩ B ⊆ carrier2" by auto
    moreover from aF bF have "a ∩ b ∈ F" by auto
    moreover from aF bF fa fb have "f `(a ∩ b) ⊆ A ∩ B" by auto
  qed

```

```

    ultimately show "A∩B ∈ fimage F" by auto
  qed auto
qed

```

```

lemma Int_filters:
  fixes F and T (structure) assumes "filter F T"
  fixes E assumes "filter E T"
  shows "F ∩ E ∈ Filters"
proof -
  interpret filter F T by fact
  interpret filter E T by fact
  show ?thesis by auto
qed

```

```

lemma ultraCI [intro!]:
  fixes T (structure)
  shows "( $\bigwedge A. [A \subseteq \text{carrier}; \text{carrier} - A \notin F] \implies A \in F$ )  $\implies$  ultra F"
  by (auto simp: ultr_def)

```

```

lemma ultraE:
  fixes T (structure)
  shows "[ultra F; A  $\subseteq$  carrier;
    A  $\in$  F  $\implies$  R;
    carrier - A  $\in$  F  $\implies$  R
  ]  $\implies$  R"
  by (auto simp: ultr_def)

```

```

lemma ultraD:
  fixes T (structure)
  shows "[ultra F; A  $\subseteq$  carrier; A  $\notin$  F]  $\implies$  (carrier - A)  $\in$  F"
  by (erule ultraE) auto

```

```

locale ultra_filter = filter +
  assumes ultra: "ultra F"
  notes ultraD = ultraD [OF ultra]
  notes ultraE [elim] = ultraE [OF ultra]

```

```

lemma (in ultra_filter) max:
  fixes E assumes "filter E T"
  assumes fsube: "F  $\subseteq$  E"
  shows "E  $\subseteq$  F"
proof -
  interpret filter E T by fact
  show ?thesis proof
    fix x assume xinE: "x  $\in$  E"
    hence "x  $\subseteq$  carrier" ..
    hence "x  $\in$  F  $\vee$  carrier - x  $\in$  F" by auto
    thus "x  $\in$  F"
  proof clarify
    assume "carrier - x  $\in$  F"
    hence "carrier - x  $\in$  E" using fsube ..
  end
end

```

```

    with xinE have "x ∩ (carrier - x) ∈ E" ..
    hence False by auto
    thus "x ∈ F" ..
  qed
qed
qed

lemma (in filter) max_ultra:
  assumes carrier_not_empty: "carrier ≠ {}"
  and fmax: "∀ E ∈ Filters. F ⊆ E ⟶ F = E"
  shows "ultra F"
proof

  fix A let ?CA = "carrier - A"
  assume A_subset_carrier: "A ⊆ carrier"
    and CA_notin_F: "?CA ∉ F"

  let ?E = "{V. ∃ U ∈ F. V ⊆ carrier ∧ A ∩ U ⊆ V}"
  have "?E ∈ Filters"
proof
  show "{} ∉ ?E"
  proof clarify
    fix U assume U_in_F: "U ∈ F" and "A ∩ U ⊆ {}"
    hence "U ⊆ ?CA" by auto
    with U_in_F have "?CA ∈ F" by auto
    with CA_notin_F show False ..
  qed
next show "⋃ ?E ⊆ carrier" by auto
next fix a b assume "a ∈ ?E" and "b ∈ ?E"
  then obtain u v where props: "u ∈ F" "a ⊆ carrier" "A ∩ u ⊆ a"
    "v ∈ F" "b ⊆ carrier" "A ∩ v ⊆ b" by auto
  hence "(u ∩ v) ∈ F" "a ∩ b ⊆ carrier" "A ∩ (u ∩ v) ⊆ a ∩ b"
    by auto
  thus "a ∩ b ∈ ?E" by auto
next fix a b assume "a ∈ ?E" and asub: "a ⊆ b" and bsub: "b ⊆ carrier"
  thus "b ∈ ?E" by blast
qed

moreover have "F ⊆ ?E" by auto

moreover from carrier_not_empty
have "{carrier} ∈ Filters" by auto
hence "F ≠ {}" using fmax by blast
hence "A ∈ ?E" using A_subset_carrier by auto

ultimately show "A ∈ F" using fmax by auto

qed

lemma filter_chain_lemma:
  fixes T (structure) and F
  assumes "filter F T"
  assumes C_chain: "C ∈ chain {V. V ∈ Filters ∧ F ⊆ V}" (is "C ∈ chain ?FF")

```

```

shows " $\bigcup (C \cup \{F\}) \in \text{Filters}$ " (is "?E  $\in$  Filters")
proof-
interpret filter F T by fact
from C_chain have C_subset_FF[dest]: " $\bigwedge x. x \in C \implies x \in ?FF$ " and
  C_ordered: " $\forall A \in C. \forall B \in C. A \subseteq B \vee B \subseteq A$ "
  by (auto simp: chain_def chain_subset_def)

show ?thesis
proof
  show "{}  $\notin$  ?E" by (auto dest: filtersD1)
next
  show " $\bigcup ?E \subseteq \text{carrier}$ " by (blast dest: filtersD2)
next
  fix a b assume a_in_E: "a  $\in$  ?E" and a_subset_b: "a  $\subseteq$  b"
  and b_subset_carrier: "b  $\subseteq$  carrier"
  thus "b  $\in$  ?E" by (blast dest: filtersD4)
next
  fix a b assume a_in_E: "a  $\in$  ?E" and b_in_E: "b  $\in$  ?E"
  then obtain A B where A_in_chain: "A  $\in C \cup \{F\}$ " and B_in_chain: "B  $\in C \cup \{F\}$ "
    and a_in_A: "a  $\in$  A" and b_in_B: "b  $\in$  B" and A_filter: "A  $\in$  Filters"
    and B_filter: "B  $\in$  Filters"
    by auto
  with C_ordered have "A  $\subseteq$  B  $\vee$  B  $\subseteq$  A" by auto
  thus "a  $\cap$  b  $\in$  ?E"
proof
  assume "A  $\subseteq$  B"
  with a_in_A have "a  $\in$  B" ..
  with B_filter b_in_B have "a  $\cap$  b  $\in$  B" by (intro filtersD3)
  with B_in_chain show ?thesis ..
next
  assume "B  $\subseteq$  A" — Symmetric case
  with b_in_B A_filter a_in_A A_in_chain
  show ?thesis by (blast intro: filtersD3)
qed
qed
qed

```

```

lemma expand_filter_ultra:
  fixes T (structure)
  assumes carrier_not_empty: "carrier  $\neq$  {}"
  and F_filter: "F  $\in$  Filters"
  and R: " $\bigwedge U. [\![ U \in \text{Filters}; F \subseteq U; \text{ultra } U ]\!] \implies R$ "
  shows "R"
proof-
let ?FF = "{V. V  $\in$  Filters  $\wedge$  F  $\subseteq$  V}"
have " $\forall C \in \text{chain } ?FF. \exists y \in ?FF. \forall x \in C. x \subseteq y$ "
proof clarify
  fix C let ?M = " $\bigcup (C \cup \{F\})$ "
  assume C_in_chain: "C  $\in$  chain ?FF"
  hence "?M  $\in$  ?FF" using F_filter
    by (auto dest: filter_chain_lemma [OF filter.intro])
  moreover have " $\forall x \in C. x \subseteq ?M$ " using C_in_chain
    by (auto simp: chain_def)

```

```

ultimately show "∃y∈?FF. ∀x∈C. x ⊆ y"
  by auto
qed then obtain U where
  U_FFfilter: "U ∈ ?FF" and U_max: "∀ V ∈ ?FF. U ⊆ V ⟶ U = V"
  by (blast dest!: Zorn_Lemma2)
hence U_filter: "U ∈ Filters" and F_subset_U: "F ⊆ U"
  by auto
moreover from U_filter carrier_not_empty have "ultra U"
proof (rule filter.max_ultra [OF filter.intro], clarify)
  fix E x assume "E ∈ Filters" and U_subset_E: "U ⊆ E" and x_in_E: "x ∈ E"
  with F_subset_U have "E ∈ ?FF" by auto
  with U_subset_E x_in_E U_max show "x ∈ U" by blast
qed
ultimately show ?thesis
  by (rule R)
qed

```

1.10 Convergence

definition

```

converges :: "'a top ⇒ 'a set set ⇒ 'a ⇒ bool" ("(_ ⟶ι _)") [55, 55] 55) where
"converges T F x ⟷ nhds T x ⊆ F"

```

notation (xsymbols)

```

converges ("(_ ⊢ _ ⟶ _)" [55, 55, 55] 55)

```

definition

```

cnvgnt :: "'a top ⇒ 'a set set ⇒ bool" ("_ convergentι" [50] 50) where
"cnvgnt T F ⟷ (∃ x ∈ carr T. converges T F x)"

```

definition

```

limites :: "'a top ⇒ 'a set set ⇒ 'a set" ("limsι") where
"limites T F = {x. x ∈ carr T ∧ T ⊢ F ⟶ x}"

```

definition

```

limes :: "'a top ⇒ 'a set set ⇒ 'a" ("limι") where
"limes T F = (THE x. x ∈ carr T ∧ T ⊢ F ⟶ x)"

```

lemma (in carrier) convergesI [intro]:

```

"nhds x ⊆ F ⟹ F ⟶ x"
by (auto simp: converges_def)

```

lemma (in carrier) convergesE [elim]:

```

"[[ F ⟶ x; nhds x ⊆ F ⟹ R ]] ⟹ R"
by (auto simp: converges_def)

```

lemma (in carrier) convergentI [intro?]:

```

"[[ F ⟶ x; x ∈ carrier ]] ⟹ F convergent"
by (auto simp: cnvgnt_def)

```

lemma (in carrier) convergentE [elim]:

```

"[[ F convergent;

```

```


$$\bigwedge x. [ F \longrightarrow x; x \in \text{carrier} ] \implies R$$

]  $\implies R$ "
by (auto simp: cnvgnt_def)

```

lemma (in continuous) fimage_converges:

```

assumes xpoint: "x ∈ carrier"
and conv: "F  $\longrightarrow_1$  x"
shows "fimage F  $\longrightarrow_2$  (f x)"

```

proof (rule, rule)

```
fix v assume vnhd: "v ∈ nhds2 (f x)"
```

```
then obtain m where v_subset_carrier: "v ⊆ carrier2"
```

```
and m_open: "m open2"
```

```
and m_subset_v: "m ⊆ v"
```

```
and fx_in_m: "f x ∈ m" ..
```

```
let ?m' = "carrier ∩ f-`m"
```

```
from fx_in_m xpoint have "x ∈ ?m'" by auto
```

```
with m_open have "?m' ∈ nhds x" by auto
```

```
with conv have "?m' ∈ F" by auto
```

```
moreover from m_subset_v have "f`?m' ⊆ v" by auto
```

```
ultimately show "v ∈ fimage F" using v_subset_carrier by auto
```

qed

corollary (in continuous) fimage_convergent [intro!]:

```
"F convergent1  $\implies$  fimage F convergent2"
```

```
by (blast intro: convergentI fimage_converges)
```

lemma (in topology) closure_convergent_filter:

```
assumes xclosure: "x ∈ closure A"
```

```
and xpoint: "x ∈ carrier"
```

```
and asub: "A ⊆ carrier"
```

```
and H: " $\bigwedge F. [ F \in \text{Filters}; F \longrightarrow x; A \in F ] \implies R$ "
```

```
shows "R"
```

proof-

```
let ?F = "{v. v ⊆ carrier ∧ (∃ u ∈ nhds x. u ∩ A ⊆ v)}"
```

```
have "?F ∈ Filters"
```

proof

```
from asub xclosure have adhx: "x adh A" by (rule closure_imp_adh)
```

```
thus "{} ∉ ?F" by (auto elim: adhCE)
```

```
next show " $\bigcup ?F \subseteq \text{carrier}$ " by auto
```

```
next fix a b assume aF: "a ∈ ?F" and bF: "b ∈ ?F"
```

```
then obtain u v where
```

```
aT: "a ⊆ carrier" and bT: "b ⊆ carrier" and
```

```
ux: "u ∈ nhds x" and vx: "v ∈ nhds x" and
```

```
uA: "u ∩ A ⊆ a" and vA: "v ∩ A ⊆ b"
```

```
by auto
```

```
moreover from ux vx have "u ∩ v ∈ nhds x"
```

```
by (auto intro: nhds_inter)
```

```
moreover from uA vA have "(u ∩ v) ∩ A ⊆ a ∩ b" by auto
```

```
ultimately show "a ∩ b ∈ ?F" by auto
```

```
next fix a b assume aF: "a ∈ ?F" and ab: "a ⊆ b" and bT: "b ⊆ carrier"
```

```
then obtain u
```

```
where at: "a ⊆ carrier" and ux: "u ∈ nhds x" and uA: "u ∩ A ⊆ a"
```

```
by auto
```

```

    moreover from ux bT have "u ∪ b ∈ nhds x"
      by (auto intro: nhds_greater)
    moreover from uA ab have "(u ∪ b) ∩ A ⊆ b" by auto
    ultimately show "b ∈ ?F" by auto
  qed
  moreover have "?F ⟶ x"
    by auto
  moreover from asub xpoint have "A ∈ ?F"
    by (blast intro: carrier_nhd)
  ultimately show ?thesis
    by (rule H)
  qed

```

```

lemma convergent_filter_closure:
  fixes F and T (structure)
  assumes "filter F T"
  assumes converge: "F ⟶ x"
  and xpoint: "x ∈ carrier"
  and AF: "A ∈ F"
  shows "x ∈ closure A"

```

```

proof-
  interpret filter F T by fact
  have "x adh A"
  proof
    fix u assume unhd: "u ∈ nhds x"
    with converge have "u ∈ F" by auto
    with AF have "u ∩ A ∈ F" by auto
    thus "u ∩ A ≠ {}" by blast
  qed
  with xpoint show ?thesis
    by (rule adh_imp_closure)
  qed

```

1.11 Separation

1.11.1 T₀ spaces

```

locale T0 = topology +
  assumes T0: "∀ x ∈ carrier. ∀ y ∈ carrier. x ≠ y ⟶
    (∃ u ∈ nhds x. y ∉ u) ∨ (∃ v ∈ nhds y. x ∉ v)"

```

```

lemma (in T0) T0_eqI:
  assumes points: "x ∈ carrier" "y ∈ carrier"
  and R1: "⋀ u. u ∈ nhds x ⟹ y ∈ u"
  and R2: "⋀ v. v ∈ nhds y ⟹ x ∈ v"
  shows "x = y"
  using T0 points
  by (auto intro: R1 R2)

```

```

lemma (in T0) T0_neqE [elim]:

```

```

assumes x_neq_y: "x ≠ y"
and points: "x ∈ carrier" "y ∈ carrier"
and R1: "∧u. [ u ∈ nhds x; y ∉ u ] ⇒ R"
and R2: "∧v. [ v ∈ nhds y; x ∉ v ] ⇒ R"
shows "R"
using T0 points x_neq_y
by (auto intro: R1 R2)

```

1.11.2 T1 spaces

```

locale T1 = T0 +
  assumes DT01: "∀ x ∈ carrier. ∀ y ∈ carrier. x ≠ y →
    (∃ u ∈ nhds x. y ∉ u) = (∃ v ∈ nhds y. x ∉ v)"

```

```

lemma (in T1) T1_neqE [elim]:
  assumes x_neq_y: "x ≠ y"
  and points: "x ∈ carrier" "y ∈ carrier"
  and R [intro] : "∧u v. [ u ∈ nhds x; v ∈ nhds y; y ∉ u; x ∉ v ] ⇒ R"
  shows "R"

```

```

proof-
  from DT01 x_neq_y points
  have nhd_iff: "(∃ v ∈ nhds y. x ∉ v) = (∃ u ∈ nhds x. y ∉ u)"
  by force
  from x_neq_y points show ?thesis
  proof
    fix u assume u_nhd: "u ∈ nhds x" and y_notin_u: "y ∉ u"
    with nhd_iff obtain v where "v ∈ nhds y" and "x ∉ v" by blast
    with u_nhd y_notin_u show "R" by auto
  next
    fix v assume v_nhd: "v ∈ nhds y" and x_notin_v: "x ∉ v"
    with nhd_iff obtain u where "u ∈ nhds x" and "y ∉ u" by blast
    with v_nhd x_notin_v show "R" by auto
  qed
qed

```

```

declare (in T1) T0_neqE [rule del]

```

```

lemma (in T1) T1_eqI:
  assumes points: "x ∈ carrier" "y ∈ carrier"
  and R1: "∧u v. [ u ∈ nhds x; v ∈ nhds y; y ∉ u ] ⇒ x ∈ v"
  shows "x = y"
proof (rule ccontr)
  assume "x ≠ y" thus False using points
  by (auto intro: R1)
qed

```

```

lemma (in T1) singleton_closed [iff]: "{x} closed"
proof (cases "x ∈ carrier")
  case False hence "carrier - {x} = carrier"
  by auto
  thus ?thesis by (clarsimp intro!: closedI)
next

```

```

case True show ?thesis
proof (rule closedI, rule open_kriterion)
  fix y assume "y ∈ carrier - {x}"
  hence "y ∈ carrier" "x ≠ y" by auto
  with True obtain v where "v ∈ nhds y" "x ∉ v"
    by (elim T1_neqE)
  then obtain m where "m open" "y ∈ m" "m ⊆ carrier - {x}"
    by (auto elim!: nhdE)
  thus "∃ m. m open ∧ y ∈ m ∧ m ⊆ carrier - {x}"
    by blast
qed
qed

```

```

lemma (in T1) finite_closed:
  assumes finite: "finite A"
  shows "A closed"
  using finite
proof induct
  case empty show ?case ..
next
  case (insert x F)
  hence "{x} ∪ F closed" by blast
  thus ?case by simp
qed

```

1.11.3 T2 spaces (Hausdorff spaces)

```

locale T2 = T1 +
  assumes T2: "∀ x ∈ carrier. ∀ y ∈ carrier. x ≠ y
    → (∃ u ∈ nhds x. ∃ v ∈ nhds y. u ∩ v = {})"

```

```

lemma T2_axiomsI:
  fixes T (structure)
  shows
    "(∀ x y. [ x ∈ carrier; y ∈ carrier; x ≠ y ] ⇒
      ∃ u ∈ nhds x. ∃ v ∈ nhds y. u ∩ v = {})
    ⇒ T2_axioms T"
  by (auto simp: T2_axioms_def)

```

```

declare (in T2) T1_neqE [rule del]

```

```

lemma (in T2) neqE [elim]:
  assumes neq: "x ≠ y"
  and points: "x ∈ carrier" "y ∈ carrier"
  and R: "∧ u v. [ u ∈ nhds x; v ∈ nhds y; u ∩ v = {} ] ⇒ R"
  shows R
proof-
  from T2 points neq obtain u v where
    "u ∈ nhds x" "v ∈ nhds y" "u ∩ v = {}" by force
  thus R by (rule R)
qed

```

```

lemma (in T2) neqE2 [elim]:

```

```

assumes neq: "x ≠ y"
and points: "x ∈ carrier" "y ∈ carrier"
and R: "∧ u v. [ u ∈ nhds x; v ∈ nhds y; z ∉ u ∨ z ∉ v ] ⇒ R"
shows R
proof-
from T2 points neq obtain u v where
  "u ∈ nhds x" "v ∈ nhds y" "u ∩ v = {}" by force
thus R by (auto intro: R)
qed

```

```

lemma T2_axiom_implies_T1_axiom:
  fixes T (structure)
  assumes T2: "∀ x ∈ carrier. ∀ y ∈ carrier. x ≠ y
    → (∃ u ∈ nhds x. ∃ v ∈ nhds y. u ∩ v = {})"
  shows "T1_axioms T"
proof (rule T1_axioms.intro, clarify)
  interpret carrier T .
  fix x y assume neq: "x ≠ y" and
    points: "x ∈ carrier" "y ∈ carrier"
  with T2 obtain u v
    where unhd: "u ∈ nhds x" and
      vnhd: "v ∈ nhds y" and Int_empty: "u ∩ v = {}"
    by force
  show "(∃ u ∈ nhds x. y ∉ u) = (∃ v ∈ nhds y. x ∉ v)"
  proof safe
    show "∃ v ∈ nhds y. x ∉ v"
    proof
      from unhd have "x ∈ u" by auto
      with Int_empty show "x ∉ v" by auto
    qed (rule vnhd)
  next
    show "∃ u ∈ nhds x. y ∉ u"
    proof
      from vnhd have "y ∈ v" by auto
      with Int_empty show "y ∉ u" by auto
    qed (rule unhd)
  qed
qed

```

```

lemma T2_axiom_implies_T0_axiom:
  fixes T (structure)
  assumes T2: "∀ x ∈ carrier. ∀ y ∈ carrier. x ≠ y
    → (∃ u ∈ nhds x. ∃ v ∈ nhds y. u ∩ v = {})"
  shows "T0_axioms T"
proof (rule T0_axioms.intro, clarify)
  interpret carrier T .
  fix x y assume neq: "x ≠ y" and
    points: "x ∈ carrier" "y ∈ carrier"
  with T2 obtain u v
    where unhd: "u ∈ nhds x" and
      vnhd: "v ∈ nhds y" and Int_empty: "u ∩ v = {}"
    by force
  show "∃ u ∈ nhds x. y ∉ u"

```

```

proof
  from vnhd have "y ∈ v" by auto
  with Int_empty show "y ∉ u" by auto
qed (rule unhd)
qed

lemma T2I:
  fixes T (structure) assumes "topology T"
  assumes I: "∧x y. [ x ∈ carrier; y ∈ carrier; x ≠ y ] ⇒
             ∃ u ∈ nhds x. ∃ v ∈ nhds y. u ∩ v = {}"
  shows "T2 T"
proof -
  interpret topology T by fact
  show ?thesis apply intro_locales
    apply (rule T2_axiom_implies_T0_axiom)
    using I apply simp
    apply (rule T2_axiom_implies_T1_axiom)
    using I apply simp
    apply unfold_locales
    using I apply simp
  done
qed

lemmas T2E = T2.neqE
lemmas T2E2 = T2.neqE2

lemma (in T2) unique_convergence:
  fixes F assumes "filter F T"
  assumes points: "x ∈ carrier" "y ∈ carrier"
    and Fx: "F ⟶ x"
    and Fy: "F ⟶ y"
  shows "x = y"
proof -
  interpret filter F T by fact
  show ?thesis proof (rule ccontr)
    assume "x ≠ y" then obtain u v
      where unhdx: "u ∈ nhds x"
        and vnhdv: "v ∈ nhds y"
        and inter: "u ∩ v = {}"
      using points ..
    hence "u ∈ F" and "v ∈ F" using Fx Fy by auto
    hence "u ∩ v ∈ F" ..
    with inter show "False" by auto
  qed
qed

lemma (in topology) unique_convergence_implies_T2 [rule_format]:
  assumes unique_convergence:
    "∧x y F. [ x ∈ carrier; y ∈ carrier; F ∈ Filters;
              F ⟶ x; F ⟶ y ] ⇒ x = y"
  shows "T2 T"

```

```

proof (rule T2I)
  show "topology T" ..
next
fix x y assume points: "x ∈ carrier" "y ∈ carrier"
  and neq: "x ≠ y"
show "∃u∈nhds x. ∃v∈nhds y. u ∩ v = {}"
proof (rule ccontr, simp)
  assume non_empty_Int: "∀u∈nhds x. ∀v∈nhds y. u ∩ v ≠ {}"
  let ?E = "{w. w ⊆ carrier ∧ (∃ u ∈ nhds x. ∃ v ∈ nhds y. u ∩ v ⊆ w)}"

  have "?E ∈ Filters"
  proof rule
    show "{} ∉ ?E" using non_empty_Int by auto
  next show "⋃?E ⊆ carrier" by auto
  next fix a b assume "a ∈ ?E" "b ∈ ?E"
    then obtain ua va ub vb
      where "a ⊆ carrier" "ua ∈ nhds x" "va ∈ nhds y" "ua ∩ va ⊆ a"
        "b ⊆ carrier" "ub ∈ nhds x" "vb ∈ nhds y" "ub ∩ vb ⊆ b"
      by auto
    hence "a ∩ b ⊆ carrier" "ua ∩ ub ∈ nhds x" "va ∩ vb ∈ nhds y" "(ua ∩ ub) ∩ (va
  ∩ vb) ⊆ a ∩ b"
      by (auto intro!: nhds_inter simp: Int_ac)
    thus "a ∩ b ∈ ?E" by blast
  next fix a b assume "a ∈ ?E" and a_sub_b:
    "a ⊆ b" and b_sub_carrier: "b ⊆ carrier"
    then obtain u v
      where u_int_v: "u ∩ v ⊆ a" and nhds: "u ∈ nhds x" "v ∈ nhds y"
      by auto
    from u_int_v a_sub_b have "u ∩ v ⊆ b" by auto
    with b_sub_carrier nhds show "b ∈ ?E" by blast
  qed

  moreover have "?E ⟶ x"
  proof (rule, rule)
    fix w assume "w ∈ nhds x"
    moreover have "carrier ∈ nhds y" using 'y ∈ carrier' ..
    moreover have "w ∩ carrier ⊆ w" by auto
    ultimately show "w ∈ ?E" by auto
  qed

  moreover have "?E ⟶ y"
  proof (rule, rule)
    fix w assume "w ∈ nhds y"
    moreover have "carrier ∈ nhds x" using 'x ∈ carrier' ..
    moreover have "w ∩ carrier ⊆ w" by auto
    ultimately show "w ∈ ?E" by auto
  qed

  ultimately have "x = y" using points
  by (auto intro: unique_convergence)
  thus False using neq by contradiction
qed
qed

```

```

lemma (in T2) limI [simp]:
  assumes filter: "F ∈ Filters"
  and      point: "x ∈ carrier"
  and     converges: "F -> x"
  shows "lim F = x"
  using filter converges point
  by (auto simp: limes_def dest: unique_convergence [OF filter.intro])

```

```

lemma (in T2) convergent_limE:
  assumes convergent: "F convergent"
  and filter: "F ∈ Filters"
  and R: "[ lim F ∈ carrier; F -> lim F ] ==> R"
  shows "R"
  using convergent filter
  by (force intro!: R)

```

```

lemma image_lim_subset_lim_fimage:
  fixes f and S (structure) and T (structure) and fimage
  defines "fimage ≡ fimg T f"
  assumes "continuous f S T"
  shows "F ∈ Filters1 ==> f'(lims F) ⊆ lims2 (fimage F)"
proof -
  interpret continuous f S T fimage by fact fact
  show ?thesis by (auto simp: limites_def intro: fimage_converges)
qed

```

1.11.4 T3 axiom and regular spaces

```

locale T3 = topology +
  assumes T3: "∀ A. ∀ x ∈ carrier - A. A ⊆ carrier ∧ A closed →
              (∃ B. ∃ U ∈ nhds x. B open ∧ A ⊆ B ∧ B ∩ U = {})"

```

```

lemma (in T3) T3E:
  assumes H: "A ⊆ carrier" "A closed" "x ∈ carrier" "x ∉ A"
  and      R: "∧ B U. [ A ⊆ B; B open; U ∈ nhds x; B ∩ U = {} ] ==> R"
  shows "R"
  using T3 H
  by (blast dest: R)

```

```

locale regular = T1 + T3

```

```

lemma regular_implies_T2:
  fixes T (structure)
  assumes "regular T"
  shows "T2 T"
proof (rule T2I)
  interpret regular T by fact
  show "topology T" ..
next
  interpret regular T by fact
  fix x y assume "x ∈ carrier" "y ∈ carrier" "x ≠ y"
  hence "{y} ⊆ carrier" "{y} closed" "x ∈ carrier" "x ∉ {y}" by auto

```

```

then obtain B U where B: "{y} ⊆ B" "B open" and U: "U ∈ nhds x" "B ∩ U = {}"
  by (elim T3E)
from B have "B ∈ nhds y" by auto
thus "∃u∈nhds x. ∃v∈nhds y. u ∩ v = {}" using U
  by blast
qed

```

1.11.5 T4 axiom and normal spaces

```

locale T4 = topology +
  assumes T4: "∀ A B. A closed ∧ A ⊆ carrier ∧ B closed ∧ B ⊆ carrier ∧
    A ∩ B = {} → (∃ U V. U open ∧ A ⊆ U ∧ V open ∧ B ⊆ V ∧ U ∩ V = {})"

```

lemma (in T4) T4E:

```

  assumes H: "A closed" "A ⊆ carrier" "B closed" "B ⊆ carrier" "A ∩ B = {}"
  and R: "∧ U V. [ U open; A ⊆ U; V open; B ⊆ V; U ∩ V = {} ] ⇒ R"
  shows "R"

```

proof-

```

  from H T4 have "(∃ U V. U open ∧ A ⊆ U ∧ V open ∧ B ⊆ V ∧ U ∩ V = {})"
    by auto
  then obtain U V where "U open" "A ⊆ U" "V open" "B ⊆ V" "U ∩ V = {}"
    by auto
  thus ?thesis by (rule R)

```

qed

locale normal = T1 + T4

lemma normal_implies_regular:

```

  fixes T (structure)
  assumes "normal T"
  shows "regular T"

```

proof -

```

  interpret normal T by fact
  show ?thesis
  proof intro_locales
    show "T3_axioms T"
    proof (rule T3_axioms.intro, clarify)
      fix A x assume x: "x ∈ carrier" "x ∉ A" and A: "A closed" "A ⊆ carrier"
      from x have "{x} closed" "{x} ⊆ carrier" "A ∩ {x} = {}" by auto
      with A obtain U V
        where "U open" "A ⊆ U" "V open" "{x} ⊆ V" "U ∩ V = {}" by (rule T4E)
      thus "∃B. ∃U∈nhds x. B open ∧ A ⊆ B ∧ B ∩ U = {}" by auto
    qed
  qed

```

qed

qed

qed

end

2 The topology of llists

```
theory LList_Topology
imports Topology "../Lazy-Lists-II/LList2"
begin
```

2.1 The topology of all llists

This theory introduces the topologies of all llists, of infinite llists, and of the non-empty llists. For all three cases it is proved that safety properties are closed sets and that liveness properties are dense sets. Finally, we prove in each of the the three different topologies the respective theorem of Alpern and Schneider [1], which states that every property can be represented as an intersection of a safety property and a liveness property.

definition

```
ttop :: "'a set  $\Rightarrow$  'a llist top" where
"ttop A = topo ( $\bigcup$  s $\in$ A*. {suff A s})"
```

```
lemma ttop_topology [iff]: "topology (ttop A)"
  by (auto simp: ttop_def)
```

```
locale suffixes =
  fixes A and B
  defines [simp]: "B  $\equiv$  ( $\bigcup$  s $\in$ A*. {suff A s})"
```

```
locale trace_top = suffixes + topobase
```

```
lemma (in trace_top) open_iff [iff]:
  "m open = (m  $\in$  topo ( $\bigcup$  s $\in$ A*. {suff A s}))"
  by (simp add: T_def is_open_def)
```

```
lemma (in trace_top) suff_open [intro!]:
  "r  $\in$  A*  $\implies$  suff A r open"
  by auto
```

```
lemma (in trace_top) ttop_carrier: "A $^\infty$  = carrier"
  by (auto simp: carrier_topo B_def suff_def)
```

```
lemma (in trace_top) suff_nhd_base:
  assumes unhd: "u  $\in$  nhds t"
  and H: " $\bigwedge$ r. [ r  $\in$  finpref A t; suff A r  $\subseteq$  u ]  $\implies$  R"
  shows "R"
```

proof-

```
  from unhd obtain m where
    uA: "u  $\subseteq$  A $^\infty$ " and
    mopen: "m open" and
    tm: "t  $\in$  m" and
    mu: "m  $\subseteq$  u"
  by (auto simp: ttop_carrier [THEN sym])
  from mopen tm have
    " $\exists$ r  $\in$  finpref A t. suff A r  $\subseteq$  m"
```

```

proof (induct "m")
  case (basic a)
  then obtain s where sA: "s ∈ A*" and as: "a = suff A s" and ta: "t ∈ a"
    by auto
  from sA as ta have "s ∈ finpref A t" by (auto dest: suff_finpref)
  thus ?case using as by auto
next case (inter a b)
  then obtain r r' where
    rt: "r ∈ finpref A t" and ra: "suff A r ⊆ a" and
    r't: "r' ∈ finpref A t" and r'b: "suff A r' ⊆ b"
    by auto
  from rt r't have "r ≤ r' ∨ r' ≤ r"
    by (auto simp: finpref_def dest: pref_locally_linear)
  thus ?case
  proof
    assume "r ≤ r'"
    hence "suff A r' ⊆ suff A r" by (rule suff_mono2)
    thus ?case using r't ra r'b by auto
  next assume "r' ≤ r"
    hence "suff A r ⊆ suff A r'" by (rule suff_mono2)
    thus ?case using rt r'b ra by auto
  qed
next case (union M)
  then obtain v where
    "t ∈ v" and vM: "v ∈ M"
    by blast
  then obtain r where "r ∈ finpref A t" "suff A r ⊆ v" using union
    by auto
  thus ?case using vM by auto
qed
with mu show ?thesis by (auto intro: H)
qed

lemma (in trace_top) nhds_LNil [simp]: "nhds LNil = {A∞}"
proof
  show "nhds LNil ⊆ {A∞}"
  proof clarify
    fix x assume xnhd: "x ∈ nhds LNil"
    then obtain r
      where rfinpref: "r ∈ finpref A LNil" and suffsub: "suff A r ⊆ x"
      by (rule suff_nhd_base)
    from rfinpref have "r = LNil" by auto
    hence "suff A r = A∞" by auto
    with suffsub have "A∞ ⊆ x" by auto
    moreover from xnhd have "x ⊆ A∞" by (auto simp: ttop_carrier elim!: nhdE)
    ultimately show "x = A∞" by auto
  qed
next
  show "{A∞} ⊆ nhds LNil" by (auto simp: ttop_carrier)
qed

lemma (in trace_top) adh_lemma:
  assumes xpoint: "x ∈ A∞"

```

```

and PA: "P ⊆ A∞"
shows "(x adh P) = (∀ r ∈ finpref A x. ∃ s ∈ A∞. r @@ s ∈ P)"
proof-
  from PA have "∧r. r ∈ A* ⇒ (∃ s ∈ A∞. r @@ s ∈ P) =
    (∃ s ∈ P. r ≤ s)"
  by (auto simp: llist_le_def iff: lapp_allT_iff)
  hence "(∀ r ∈ finpref A x. ∃ s ∈ A∞. r @@ s ∈ P) =
    (∀ r ∈ finpref A x. ∃ s ∈ P. r ≤ s)"
  by (auto simp: finpref_def)
  also have "... = (∀ r ∈ finpref A x. suff A r ∩ P ≠ {})"
  proof-
    have "∧r. (∃ s ∈ P. r ≤ s) = ({ra. ra ∈ A∞ ∧ r ≤ ra} ∩ P ≠ {})" using PA
    by blast
    thus ?thesis by (simp add: suff_def)
  qed
  also have "... = (∀ u ∈ nhds x. u ∩ P ≠ {})"
  proof safe
    fix r assume uP: "∀ u ∈ nhds x. u ∩ P ≠ {}" and
      rfinpref: "r ∈ finpref A x" and rP: "suff A r ∩ P = {}"
    from rfinpref have "suff A r open" by (auto dest!: finpref_fin)
    hence "suff A r ∈ nhds x" using xpoint rfinpref
    by auto
    with uP rP show "False" by auto
  next
    fix u assume
      inter: "∀ r ∈ finpref A x. suff A r ∩ P ≠ {}" and
      unhd: "u ∈ nhds x" and
      uinter: "u ∩ P = {}"
    from unhd obtain r where
      "r ∈ finpref A x" and "suff A r ⊆ u"
    by (rule suff_nhd_base)
    with inter uinter show "False" by auto
  qed
  finally show ?thesis by (simp add: adhs_def)
qed

lemma (in trace_top) topology [iff]:
  "topology T"
by (simp add: T_def)

lemma (in trace_top) safety_closed_iff:
  "P ⊆ A∞ ⇒ safety A P = (P closed)"
by (auto simp: safety_def topology.closed_adh adh_lemma ttop_carrier)

lemma (in trace_top) liveness_dense_iff:
  assumes P: "P ⊆ A∞"
  shows "liveness A P = (P dense)"
proof-
  have "liveness A P = (∀ r ∈ A*. ∃ s ∈ A∞. r @@ s ∈ P)"
  by (simp add: liveness_def)
  also have "... = (∀ x ∈ A∞. ∀ r ∈ finpref A x. ∃ s ∈ A∞. r @@ s ∈ P)"
  by (auto simp: finpref_def dest: finsubsetall)
  also have "... = (∀ x ∈ A∞. x adh P)" using P

```

```

    by (simp add: adh_lemma)
  also have "... = (A∞ ⊆ closure P)" using P
    by (auto simp: adh_closure_iff ttop_carrier)
  also have "... = (P dense)"
    by (simp add: liveness_def is_dense_def is_densein_def ttop_carrier)
  finally show ?thesis .
qed

```

```

lemma (in trace_top) LNil_safety: "safety A {LNil}"
proof (unfold safety_def, clarify)
  fix t
  assume adh: "t ∈ A∞" "∀r∈finpref A t. ∃s∈A∞. r @@ s ∈ {LNil}"
  thus "t = LNil" by (cases t)(auto simp: finpref_def)
qed

```

```

lemma (in trace_top) LNil_closed: "{LNil} closed"
by (auto intro: iffD1 [OF safety_closed_iff] LNil_safety)

```

```

theorem (in trace_top) alpern_schneider:
assumes Psub: "P ⊆ A∞"
  shows "∃ S L. safety A S ∧ liveness A L ∧ P = S ∩ L"
proof-
  from Psub have "P ⊆ carrier" by (simp add: ttop_carrier)
  then obtain L S where
    Lsub: "L ⊆ carrier" and
    Ssub: "S ⊆ carrier" and
    Sclosed: "S closed" and
    Ldense: "L dense" and
    Pinter: "P = S ∩ L"
  by (blast elim: topology.ex_dense_closure_interE [OF topology])
  from Ssub Sclosed have "safety A S"
  by (simp add: safety_closed_iff ttop_carrier)
  moreover from Lsub Ldense have "liveness A L"
  by (simp add: liveness_dense_iff ttop_carrier)
  ultimately show ?thesis using Pinter
  by auto
qed

```

2.2 The topology of infinite llists

```

definition
  itop :: "'a set ⇒ 'a llist top" where
  "itop A = topo (⋃ s∈A*. {infsuff A s})"

```

```

locale infsuffixes =
  fixes A and B
  defines [simp]: "B ≡ (⋃ s∈A*. {infsuff A s})"

```

```

locale itrace_top = infsuffixes + topobase

```

```

lemma (in itrace_top) open_iff [iff]:

```

```

"m open = (m ∈ topo (⋃ s∈A*. {infsuff A s}))"
by (simp add: T_def is_open_def)

lemma (in itrace_top) topology [iff]: "topology T"
  by (auto simp: T_def)

lemma (in itrace_top) infsuff_open [intro!]:
  "r ∈ A* ⇒ infsuff A r open"
  by auto

lemma (in itrace_top) itop_carrier: "carrier = Aω"
  by (auto simp: carrier_topo B_def infsuff_def)

lemma itop_sub_ttop_base:
  fixes A :: "'a ⇒ bool"
    and B :: "('a llist ⇒ bool) ⇒ bool"
    and C :: "('a llist ⇒ bool) ⇒ bool"
  defines [simp]: "B ≡ ⋃ s∈A*. {suff A s}" and [simp]: "C ≡ ⋃ s∈A*. {infsuff A s}"
  shows "C = (⋃ t∈B. {t ∩ C})"
  by (auto simp: infsuff_def suff_def)

lemma itop_sub_ttop [folded ttop_def itop_def]:
  fixes A and C and S (structure)
  defines "C ≡ ⋃ s∈A*. {infsuff A s}" and "S ≡ topo C"
  fixes B and T (structure)
  defines "B ≡ ⋃ s∈A*. {suff A s}" and "T ≡ topo B"
  shows "subtopology S T"
proof -
  interpret itrace_top A C S by fact+
  interpret trace_top A B T by fact+
  show ?thesis
    by (auto intro: itop_sub_ttop_base [THEN subtop_lemma] simp: S_def T_def)
qed

lemma (in itrace_top) infsuff_nhd_base:
  assumes unhd: "u ∈ nhds t"
  and H: "⋀ r. [ r ∈ finpref A t; infsuff A r ⊆ u ] ⇒ R"
  shows "R"
proof-
  from unhd obtain m where
    uA: "u ⊆ Aω" and
    mopen: "m open" and
    tm: "t ∈ m" and
    mu: "m ⊆ u"
  by (auto simp: itop_carrier)
  from mopen tm have
    "∃ r ∈ finpref A t. infsuff A r ⊆ m"
  proof (induct "m")
    case (basic a)
    then obtain s where sA: "s ∈ A*" and as: "a = infsuff A s" and ta: "t ∈ a"
    by auto
    from sA as ta have "s ∈ finpref A t" by (auto dest: infsuff_finpref)
    thus ?case using as by auto
  qed

```

```

next case (inter a b)
  then obtain r r' where
    rt: "r ∈ finpref A t" and ra: "infsuff A r ⊆ a" and
    r't: "r' ∈ finpref A t" and r'b: "infsuff A r' ⊆ b"
    by auto
  from rt r't have "r ≤ r' ∨ r' ≤ r"
    by (auto simp: finpref_def dest: pref_locally_linear)
  thus ?case
  proof
    assume "r ≤ r'"
    hence "infsuff A r' ⊆ infsuff A r" by (rule infsuff_mono2)
    thus ?case using r't ra r'b by auto
  next assume "r' ≤ r"
    hence "infsuff A r ⊆ infsuff A r'" by (rule infsuff_mono2)
    thus ?case using rt r'b ra by auto
  qed
next case (union M)
  then obtain v where
    "t ∈ v" and vM: "v ∈ M"
    by blast
  then obtain r where "r ∈ finpref A t" "infsuff A r ⊆ v" using union
    by auto
  thus ?case using vM by auto
  qed
with mu show ?thesis by (auto intro: H)
qed

lemma (in itrace_top) hausdorff [iff]: "T2 T"
proof(rule T2I, clarify)
  fix x y assume xpoint: "x ∈ carrier"
    and ypoint: "y ∈ carrier"
    and neq: "x ≠ y"
  from xpoint ypoint have xA: "x ∈ Aω" and yA: "y ∈ Aω"
    by (auto simp: itop_carrier)
  then obtain s where
    sA: "s ∈ A*" and sx: "s ≤ x" and sy: "¬ s ≤ y" using neq
    by (rule inf_neqE) auto
  from neq have "y ≠ x" ..
  with yA xA obtain t where
    tA: "t ∈ A*" and ty: "t ≤ y" and tx: "¬ t ≤ x"
    by (rule inf_neqE) auto
  let ?u = "infsuff A s" and ?v = "infsuff A t"
  have inter: "?u ∩ ?v = {}"
  proof (rule ccontr, auto)
    fix z assume "z ∈ ?u" and "z ∈ ?v"
    hence "s ≤ z" and "t ≤ z" by (unfold infsuff_def) auto
    hence "s ≤ t ∨ t ≤ s" by (rule pref_locally_linear)
    thus False using sx sy tx ty by (auto dest: llist_le_trans)
  qed
  moreover {
    from sA tA have "?u open" and "?v open"
      by auto
    moreover from xA yA sx ty have "x ∈ ?u" and "y ∈ ?v"

```

```

    by (auto simp: infsuff_def)
  ultimately have "infsuff A s ∈ nhds x" and
    "infsuff A t ∈ nhds y"
    by auto }
  ultimately show "∃ u ∈ nhds x. ∃ v ∈ nhds y. u ∩ v = {}"
    by auto
qed

```

corollary (in itrace_top) unique_convergence:

```

"[[ x ∈ carrier;
  y ∈ carrier;
  F ∈ Filters ;
  F → x;
  F → y ]] ⇒ x = y"
apply (rule T2.unique_convergence)
prefer 2
apply (rule filter.intro)
apply auto
done

```

lemma (in itrace_top) adh_lemma:

```

assumes xpoint: "x ∈ Aω"
  and PA: "P ⊆ Aω"
shows "x adh P = (∀ r ∈ finpref A x. ∃ s ∈ Aω. r @@ s ∈ P)"
proof-

```

```

  from PA have "∧r. r ∈ A* ⇒ (∃ s ∈ Aω. r @@ s ∈ P) =
    (∃ s ∈ P. r ≤ s)"

```

```

  by (auto simp: llist_le_def iff: lapp_infT)

```

```

  hence "(∀ r ∈ finpref A x. ∃ s ∈ Aω. r @@ s ∈ P) =
    (∀ r ∈ finpref A x. ∃ s ∈ P. r ≤ s)"

```

```

  by (auto simp: finpref_def)

```

```

  also have "... = (∀ r ∈ finpref A x. infsuff A r ∩ P ≠ {})"

```

```

proof-

```

```

  have "∧r. (∃ s ∈ P. r ≤ s) = ({ra. ra ∈ Aω ∧ r ≤ ra} ∩ P ≠ {})" using PA
  by blast

```

```

  thus ?thesis by (simp add: infsuff_def)

```

```

qed

```

```

also have "... = (∀ u ∈ nhds x. u ∩ P ≠ {})"

```

```

proof safe

```

```

  fix r assume uP: "∀ u ∈ nhds x. u ∩ P ≠ {}" and

```

```

  rfinpref: "r ∈ finpref A x" and rP: "infsuff A r ∩ P = {}"

```

```

  from rfinpref have "infsuff A r open" by (auto dest!: finpref_fin)

```

```

  hence "infsuff A r ∈ nhds x" using xpoint rfinpref

```

```

  by auto

```

```

  with uP rP show "False" by auto

```

```

next

```

```

  fix u assume

```

```

  inter: "∀ r ∈ finpref A x. infsuff A r ∩ P ≠ {}" and

```

```

  unhd: "u ∈ nhds x" and

```

```

  uinter: "u ∩ P = {}"

```

```

  from unhd obtain r where

```

```

  "r ∈ finpref A x" and "infsuff A r ⊆ u"

```

```

    by (rule infsuff_nhd_base)
    with inter uinter show "False" by auto
  qed
  finally show ?thesis by (simp add: adhs_def)
qed

lemma (in itrace_top) infsafety_closed_iff:
  "P  $\subseteq$  A $^\omega$   $\implies$  infsafety A P = (P closed)"
  by (auto simp: infsafety_def topology.closed_adh adh_lemma itop_carrier)

lemma (in itrace_top) empty:
  "A = {}  $\implies$  T = {{{}}"
proof (auto simp: T_def)
  fix m x assume "m  $\in$  topo {{{}}" and xm: "x  $\in$  m"
  thus False
    by (induct m) auto
qed

lemma itop_empty: "itop {} = {{{}}"
proof (auto simp: itop_def)
  fix m x assume "m  $\in$  topo {{{}}" and xm: "x  $\in$  m"
  thus False
    by (induct m) auto
qed

lemma infliveness_empty:
  "infliveness {} P  $\implies$  False"
  by (auto simp: infliveness_def)

lemma (in trivial) dense:
  "P dense"
  by auto

lemma (in itrace_top) infliveness_dense_iff:
  assumes notempty: "A  $\neq$  {}"
  and P: "P  $\subseteq$  A $^\omega$ "
  shows "infliveness A P = (P dense)"
proof-
  have "infliveness A P = ( $\forall r \in A^*$ .  $\exists s \in A^\omega$ . r @@ s  $\in$  P)"
    by (simp add: infliveness_def)
  also have "... = ( $\forall x \in A^\omega$ .  $\forall r \in \text{finpref A x}$ .  $\exists s \in A^\omega$ . r @@ s  $\in$  P)"
  proof-
    from notempty obtain a where "a  $\in$  A"
      by auto
    hence lc: "lconst a  $\in$  A $^\omega$ "
      by (rule lconstT)
    hence " $\bigwedge P$ . ( $\forall x \in A^\omega$ .  $\forall r \in \text{finpref A x}$ . P r) = ( $\forall r \in A^*$ . P r)"
  proof (auto dest: finpref_fin)
    fix P r assume lc: "lconst a  $\in$  A $^\omega$ "
      and Pr: " $\forall x \in A^\omega$ .  $\forall r \in \text{finpref A x}$ . P r"
      and rA: "r  $\in$  A*"
    from rA lc have rlc: "r @@ lconst a  $\in$  A $^\omega$ " by (rule lapp_fin_infT)
    moreover from rA rlc have "r  $\in$  finpref A (r @@ lconst a)"
  
```

```

      by (auto simp: finpref_def llist_le_def)
      ultimately show "P r" using Pr by auto
    qed
  thus ?thesis by simp
qed
also have "... = ( $\forall x \in A^\omega. x \text{ adh } P$ )" using P
  by (simp add: adh_lemma)
also have "... = ( $A^\omega \subseteq \text{closure } P$ )" using P
  by (auto simp: adh_closure_iff itop_carrier)
also have "... = (P dense)"
  by (simp add: infliveness_def is_dense_def is_densein_def itop_carrier)
finally show ?thesis .
qed

```

```

theorem (in itrace_top) alpern_schneider:
  assumes notempty: " $A \neq \{\}$ "
  and Psub: " $P \subseteq A^\omega$ "
  shows " $\exists S L. \text{infsafety } A S \wedge \text{infliveness } A L \wedge P = S \cap L$ "

```

```

proof-
  from Psub have "P  $\subseteq$  carrier"
  by (simp add: itop_carrier [THEN sym])
  then obtain L S where
    Lsub: "L  $\subseteq$  carrier" and
    Ssub: "S  $\subseteq$  carrier" and
    Sclosed: "S closed" and
    Ldense: "L dense" and
    Pinter: "P = S  $\cap$  L"
  by (rule topology.ex_dense_closure_interE [OF topology]) auto
  from Ssub Sclosed have "infsafety A S"
  by (simp add: infsafety_closed_iff itop_carrier)
  moreover from notempty Lsub Ldense have "infliveness A L"
  by (simp add: infliveness_dense_iff itop_carrier)
  ultimately show ?thesis using Pinter
  by auto
qed

```

2.3 The topology of non-empty llists

definition

```

ptop :: "'a set  $\Rightarrow$  'a llist top" where
"ptop A  $\equiv$  topo ( $\bigcup s \in A^\clubsuit. \{\text{suff } A s\}$ )"

```

locale possuffixes =

```

  fixes A and B
  defines [simp]: "B  $\equiv$  ( $\bigcup s \in A^\clubsuit. \{\text{suff } A s\}$ )"

```

locale ptrace_top = possuffixes + topobase

lemma (in ptrace_top) open_iff [iff]:

```

"m open = (m  $\in$  topo ( $\bigcup s \in A^\clubsuit. \{\text{suff } A s\}$ ))"
by (simp add: T_def is_open_def)

```

```

lemma (in ptrace_top) topology [iff]: "topology T"
  by (simp add: T_def)

lemma (in ptrace_top) ptop_carrier: "carrier = A♣"
by (auto simp add: carrier_topo suff_def)
  (auto elim: allsts.cases)

lemma pptop_subtop_ttop:
  fixes S (structure)
  fixes A and B and T (structure)
  defines "B ≡  $\bigcup_{s \in A^*}. \{suff\ A\ s\}"$  and "T ≡ topo B"
  defines "S ≡  $\bigcup_{t \in T}. \{t - \{LNil\}\}"$ "
  shows "subtopology S T"
by (rule subtopology.intro, auto simp add: is_open_def S_def carr_def)

lemma pptop_top:
  fixes S (structure)
  fixes A and B and T (structure)
  defines "B ≡  $\bigcup_{s \in A^*}. \{suff\ A\ s\}"$  and "T ≡ topo B"
  defines "S ≡  $\bigcup_{t \in T}. \{t - \{LNil\}\}"$ "
  shows "topology ( $\bigcup_{t \in T}. \{t - \{LNil\}\}$ )"
proof -
  interpret trace_top A B T by fact+
  show ?thesis
    by (auto intro!: subtopology.subtop_topology [OF pptop_subtop_ttop]
        trace_top.topology simp: T_def)
qed

lemma (in ptrace_top) suff_open [intro!]:
  "r ∈ A♣ ⇒ suff A r open"
  by auto

lemma (in ptrace_top) suff_ptop_nhd_base:
  assumes unhd: "u ∈ nhds t"
  and H: " $\bigwedge r. [r \in pfinpref\ A\ t; suff\ A\ r \subseteq u] \Rightarrow R$ "
  shows "R"
proof-
  from unhd obtain m where
    uA: "u ⊆ A♣" and
    mopen: "m open" and
    tm: "t ∈ m" and
    mu: "m ⊆ u"
  by (auto simp: ptop_carrier)
  from mopen tm have
    "∃ r ∈ pfinpref A t. suff A r ⊆ m"
  proof (induct "m")
    case (basic a)
    then obtain s where sA: "s ∈ A♣" and as: "a = suff A s" and ta: "t ∈ a"
    by auto
    from sA as ta have "s ∈ pfinpref A t"
    by (auto simp: pfinpref_def dest: suff_finpref)
  
```

```

    thus ?case using as by auto
next case (inter a b)
  then obtain r r' where
    rt: "r ∈ pfinpref A t" and ra: "suff A r ⊆ a" and
    r't: "r' ∈ pfinpref A t" and r'b: "suff A r' ⊆ b"
  by auto
  from rt r't have "r ≤ r' ∨ r' ≤ r"
  by (auto simp: pfinpref_def finpref_def dest: pref_locally_linear)
  thus ?case
  proof
    assume "r ≤ r'"
    hence "suff A r' ⊆ suff A r" by (rule suff_mono2)
    thus ?case using r't ra r'b by auto
  next assume "r' ≤ r"
    hence "suff A r ⊆ suff A r'" by (rule suff_mono2)
    thus ?case using rt r'b ra by auto
  qed
next case (union M)
  then obtain v where
    "t ∈ v" and vM: "v ∈ M"
  by blast
  then obtain r where "r ∈ pfinpref A t" "suff A r ⊆ v" using union
  by auto
  thus ?case using vM by auto
  qed
with mu show ?thesis by (auto intro: H)
qed

lemma pfinpref_LNil [simp]: "pfinpref A LNil = {}"
  by (auto simp: pfinpref_def)

lemma (in ptrace_top) adh_lemma:
  assumes xpoint: "x ∈ A♣"
  and P_subset_A: "P ⊆ A♣"
  shows "x adh P = (∀ r ∈ pfinpref A x. ∃ s ∈ A∞. r @@ s ∈ P)"
  proof
    assume adh_x: "x adh P"
    show "∀ r ∈ pfinpref A x. ∃ s ∈ A∞. r @@ s ∈ P"
    proof
      fix r let ?u = "suff A r"
      assume r_pfinpref_x: "r ∈ pfinpref A x"
      hence r_pos: "r ∈ A♣" by (auto dest: finpref_fin)
      hence "?u open" by auto
      hence "?u ∈ nhds x" using xpoint r_pfinpref_x
      by auto
      with adh_x have "?u ∩ P ≠ {}" by (auto elim!: adhCE)
      then obtain t where tu: "t ∈ ?u" and tP: "t ∈ P"
      by auto
      from tu obtain s where "t = r @@ s" using r_pos
      by (auto elim!: suff_appE)
      with tP show "∃ s ∈ A∞. r @@ s ∈ P" using P_subset_A r_pos
      by (auto iff: lapp_allT_iff)
    qed
  qed

```

```

next
  assume H: "∀r∈pfinpref A x. ∃s∈A∞. r @@ s ∈ P"
  show "x adh P"
  proof
    fix U assume unhd: "U ∈ nhds x"
    then obtain r where r_pfinpref_x: "r ∈ pfinpref A x" and
      suff_subset_U: "suff A r ⊆ U" by (elim suff_ptop_nhd_base)
    from r_pfinpref_x have rpos: "r ∈ A♣" by (auto intro: finpref_fin)
    show "U ∩ P ≠ {}" using rpos
    proof (cases r)
      case (LCons a l)
        hence r_pfinpref_x: "r ∈ pfinpref A x" using r_pfinpref_x
          by auto
        with H obtain s where sA: "s ∈ A∞" and asP: "r@@s ∈ P"
          by auto
        moreover have "r @@ s ∈ suff A r" using sA rpos
          by (auto simp: suff_def iff: lapp_allT_iff)
        ultimately show ?thesis using suff_subset_U by auto
      qed
    qed
  qed

lemma (in ptrace_top) possafety_closed_iff:
  "P ⊆ A♣ ⇒ possafety A P = (P closed)"
  by (auto simp: possafety_def topology.closed_adh ptop_carrier adh_lemma)

lemma (in ptrace_top) posliveness_dense_iff:
  assumes P: "P ⊆ A♣"
  shows "posliveness A P = (P dense)"
  proof-
    have "posliveness A P = (∀r∈A♣. ∃ s ∈ A∞. r @@ s ∈ P)"
      by (simp add: posliveness_def)
    also have "... = (∀x∈A♣. ∀ r ∈ pfinpref A x. ∃ s ∈ A∞. r @@ s ∈ P)"
      by (auto simp: pfinpref_def finpref_def dest: finsubsetall)
    also have "... = (∀x∈A♣. x adh P)" using P
      by (auto simp: adh_lemma simp del: poslststs_iff)
    also have "... = (A♣ ⊆ closure P)" using P
      by (auto simp: adh_closure_iff ptop_carrier simp del: poslststs_iff)
    also have "... = (P dense)"
      by (simp add: posliveness_def is_dense_def is_densein_def ptop_carrier)
    finally show ?thesis .
  qed

theorem (in ptrace_top) alpern_schneider:
  assumes Psub: "P ⊆ A♣"
  shows "∃ S L. possafety A S ∧ posliveness A L ∧ P = S ∩ L"
  proof-
    from Psub have "P ⊆ carrier" by (simp add: ptop_carrier)
    then obtain L S where
      Lsub: "L ⊆ carrier" and
      Ssub: "S ⊆ carrier" and
      Sclosed: "S closed" and

```

```

    Ldense: "L dense" and
    Pinter: "P = S  $\cap$  L"
    by (blast elim: topology.ex_dense_closure_interE [OF topology])
from Ssub Sclosed have "possafety A S"
    by (simp add: possafety_closed_iff ptop_carrier)
moreover from Lsub Ldense have "posliveness A L"
    by (simp add: posliveness_dense_iff ptop_carrier)
ultimately show ?thesis using Pinter
    by auto
qed
end

```

References

- [1] B. Alpern and F. B. Schneider. Defining Liveness. *Information Processing Letters*, 21(4):181–185, Oct. 1985.
- [2] G. McCarty. *Topology : an introduction with application to topological groups*. International series in pure and applied mathematics. Graw-Hill, New York, 1967.
- [3] B. von Querenburg. *Mengentheoretische Topologie*. Springer, Heidelberg, 3. edition, 2001.